

# HeLP

## Hertweck-Luthar-Passi method.

### 2.1

30/09/2015, The Lodge

**Andreas Bächle**

**Leo Margolis**

**Andreas Bächle**

Email: [ABachle@vub.ac.be](mailto:ABachle@vub.ac.be)

Homepage: <http://homepages.vub.ac.be/~abachle/>

Address: Vrije Universiteit Brussel  
Vakgroep Wiskunde  
Pleinlaan 2  
1050 Brussels  
Belgium

**Leo Margolis**

Email: [Leo.Margolis@mathematik.uni-stuttgart.de](mailto:Leo.Margolis@mathematik.uni-stuttgart.de)

Homepage: <http://www.igt.uni-stuttgart.de/LstDiffgeo/Margolis/>

Address: Institut für Algebra und Zahlentheorie  
Universität Stuttgart  
Pfaffenwaldring 57  
D-70569 Stuttgart  
Germany

## Copyright

© 2015 by Andreas Bächle and Leo Margolis

This package is free software and may be distributed under the terms and conditions of the GNU Public License Version 2.

## Acknowledgements

The authors are grateful to Sebastian Gutsche for endowing GAP with the 4ti2-Interface. We also would like to thank him for many very helpful discussions. We also want to give credits to the developers of the software 4ti2. Thanks go to David Avis for writing lrslib and answering our questions about it. We moreover thank Wolfgang Kimmerle for introducing us to the beautiful world of group rings. The development of this package was partially supported by the Research Foundation Flanders (FWO - Vlaanderen) and the DFG priority program SPP 1489 Algorithmic and Experimental Methods in Algebra, Geometry, and Number Theory.

# Contents

<b>1</b>	<b>Introduction</b>	<b>5</b>
<b>2</b>	<b>The main functions</b>	<b>6</b>
2.1	Zassenhaus Conjecture . . . . .	6
2.2	Prime Graph Question . . . . .	9
<b>3</b>	<b>Further functions</b>	<b>12</b>
3.1	Checks for specific orders . . . . .	12
3.2	Checks for specific orders with s-constant characters . . . . .	16
3.3	Checks for all orders . . . . .	18
3.4	Changing the used Character Table . . . . .	19
3.5	Influencing how the Systems of Inequalities are solved . . . . .	20
3.6	Checking solutions, calculating and checking solutions . . . . .	21
3.7	The Wagner test . . . . .	23
3.8	Output . . . . .	24
3.9	Eigenvalue multiplicities and character values . . . . .	25
<b>4</b>	<b>Extended examples</b>	<b>27</b>
4.1	The Character Table Library . . . . .	27
4.2	The behavior of the variable HeLP sol . . . . .	28
4.3	Saving time . . . . .	29
4.4	Using InfoLevels . . . . .	32
4.5	Non-standard characters . . . . .	33
4.6	A complete example: (PQ) for the MacLaughlin simple group . . . . .	35
<b>5</b>	<b>Background</b>	<b>37</b>
5.1	The Zassenhaus Conjecture and the Prime Graph Question . . . . .	37
5.2	Partial augmentations and the structure of HeLP sol . . . . .	37
5.3	The HeLP equations . . . . .	38
5.4	The Wagner test . . . . .	39
5.5	s-constant (and (s,t)-constant) characters . . . . .	39
5.6	Known results about (ZC) and (PQ) . . . . .	40
<b>6</b>	<b>Remarks on technical problems and the implementation</b>	<b>41</b>
6.1	Making the HeLP-package run . . . . .	41
6.2	Computations with big coefficients - precision problems . . . . .	42
6.3	How much 4ti2 is really there? . . . . .	42

<i>HeLP</i>	4
-------------	---

<b>References</b>	<b>45</b>
-------------------	-----------

<b>Index</b>	<b>46</b>
--------------	-----------

# Chapter 1

## Introduction

The purpose of this package is to provide functionalities to work with torsion units in (integral) group rings. It implements a method which was developed by I.S. Luthar and I.B.S. Passi in [LP89] and which was extended by M. Hertweck in [Her07]. These names also constitute the name of the method, suggested by A. Konovalov: *Hertweck-Luthar-Passi*. The theory behind the method is briefly described in Chapter 5.

The package uses the software 4ti2 [tt] and is hence only working on systems which have 4ti2 installed. For more information on 4ti2 and to download it, please visit [www.4ti2.de](http://www.4ti2.de). To interact with 4ti2 the package makes use of the 4ti2-Interface written by S. Gutsche, which in turn uses the package IO. Note that the latter package needs a C-part to be compiled; see the readme-file or the documentation of that package. The package also provides the possibility to use `redund` from the `lrslib` software [Avi], to remove redundant inequalities before solving the system, which might speed up the computations significantly. However, it is not a requirement to have `lrslib` installed. If the above mentioned preconditions are fulfilled, one just has to copy the HeLP-package in the GAP pkg-directory. Now the package can be loaded by typing `LoadPackage("HeLP")`; during a GAP-session. If the HeLP-package doesn't work properly on your computer, you might want to check Section 6.1 for some trouble shooting.

## Chapter 2

# The main functions

### 2.1 Zassenhaus Conjecture

This function checks whether the Zassenhaus Conjecture ((ZC) for short, cf. Section 5.1) can be proved using the HeLP-method with the data available in GAP.

#### 2.1.1 HeLP\_ZC

▷ `HeLP_ZC(OrdinaryCharacterTable/Group)` (function)

**Returns:** `true` if (ZC) can be solved using the given data, `false` otherwise

`HeLP_ZC` checks whether the Zassenhaus Conjecture can be solved for the given group using the HeLP method, the Wagner test and all character data available. The argument of the function can be either an ordinary character table or a group. In the second case it will first calculate the corresponding ordinary character table. If the group in question is nilpotent, the Zassenhaus Conjecture holds by a result of A. Weiss and the function will return `true` without performing any calculations.

If the group is not solvable, the function will check all orders which are divisors of the exponent of the group. If the group is solvable, it will only check the orders of group elements, as there can't be any torsion units of another order. The function will use the ordinary table and, for the primes  $p$  for which the group is not  $p$ -solvable, all  $p$ -Brauer tables which are available in GAP to produce as many constraints on the torsion units as possible. Additionally, the Wagner test is applied to the results, cf. Section 5.4. In case the information suffices to obtain a proof for the Zassenhaus Conjecture for this group the function will return `true` and `false` otherwise. The possible partial augmentations for elements of order  $k$  and all its powers will also be stored in the list entry `HeLP_sol[k]`.

The prior computed partial augmentations in `HeLP_sol` will not be used and will be overwritten. If you do not like the last fact, please use `HeLP_AllOrders` (3.3.1).

Example

```
gap> G := AlternatingGroup(5);
Alt( [ 1 .. 5 ] )
gap> HeLP_ZC(G);
true
gap> C := CharacterTable("A5");
CharacterTable( "A5" )
gap> HeLP_ZC(C);
true
gap> HeLP_sol;
[ [ [ [ 1 ] ] ], [ [ [ 1 ] ] ], [ [ [ 1 ] ] ] ],,
```

```

[ [ [ 0, 1 ] ], [ [ 1, 0 ] ] ], [ ],,,, [ ],,,,, [ ],,,,,,,,,,,,,, [ ]
]
gap> HeLP_PrintSolution();
Solutions for elements of order 2:
[ [      u ],
  [ [ "2a" ] ],
  [      --- ],
  [      [ 1 ] ] ]
Solutions for elements of order 3:
[ [      u ],
  [ [ "3a" ] ],
  [      --- ],
  [      [ 1 ] ] ]
Solutions for elements of order 5:
[ [      u ],
  [ [ "5a", "5b" ] ],
  [      --- ],
  [      [ 0, 1 ] ],
  [      [ 1, 0 ] ] ]
There are no admissible partial augmentations for elements of order 6.
There are no admissible partial augmentations for elements of order 10.
There are no admissible partial augmentations for elements of order 15.
There are no admissible partial augmentations for elements of order 30.

```

This is the classical example of Luthar and Passi to verify the Zassenhaus Conjecture for the alternating group of degree 5, cf. [LP89]. In the first call of `HeLP_ZC` this is checked using the character table computed by GAP using the given group, the second call uses the character table from the character table library. The entries of `HeLP_sol` are

- lists with entries 0 and 1 (at the spots 1, 2, 3 and 5, which correspond to torsion units that are conjugate to group elements),
- empty lists (at the spots 6, 10, 15 and 30, stating that there are no admissible partial augmentations for these orders),
- or are not bound (these orders were not checked as they don't divide the exponent of the group).

The function `HeLP_PrintSolution` (3.8.1) can be used to display the result in a pretty way.

Example

```

gap> C := CharacterTable( "A6" );
CharacterTable( "A6" )
gap> SetInfoLevel(HeLP_Info, 2);
gap> HeLP_ZC(C);
#I Checking order 2.
#I Checking order 3.
#I Checking order 4.
#I Checking order 5.
#I Checking order 6.
#I Checking order 10.
#I Checking order 12.
#I Checking order 15.
#I Checking order 20.

```

```

#I Checking order 30.
#I Checking order 60.
#I ZC can't be solved, using the given data, for the orders: [ 6 ].
false
gap> HeLP_sol[6];
[ [ [ 1 ], [ 0, 1 ], [ -2, 2, 1 ] ], [ [ 1 ], [ 1, 0 ], [ -2, 1, 2 ] ] ]
gap> HeLP_PrintSolution(6);
Solutions for elements of order 6:
[ [          u^3,          u^2,          u ],
  [ [ "2a" ], [ "3a", "3b" ], [ "2a", "3a", "3b" ] ],
  [          ---,          ---,          --- ],
  [ [ 1 ], [ 0, 1 ], [ -2, 2, 1 ] ],
  [ [ 1 ], [ 1, 0 ], [ -2, 1, 2 ] ] ]
gap> SetInfoLevel(HeLP_Info, 1);

```

This is the example M. Hertweck deals with in his article [Her08c]. The HeLP-method is not sufficient to verify the Zassenhaus Conjecture for this group. There are two tuples of possible partial augmentations for torsion units of order 6 which are admissible by the HeLP method. M. Hertweck used a different argument to eliminate these possibilities.

Example

```

gap> G := SmallGroup(48,30);;
gap> StructureDescription(G);
"A4 : C4"
gap> HeLP_ZC(G);
#I ZC can't be solved, using the given data, for the orders: [ 4 ].
false
gap> Size(HeLP_sol[4]);
20

```

The group `SmallGroup(48,30)` is the smallest group for which the HeLP method does not suffice to prove the Zassenhaus Conjecture. However (ZC) was proved for this group in [HK06], Proposition 4.2.

Example

```

gap> G := SmallGroup(96,65);;
gap> HeLP_ZC(G);
#I (ZC) can't be solved, using the given data, for the orders: [ 8 ].
false
gap> Size(HeLP_sol[8]);
40
gap> G := SmallGroup(160,13);;
gap> HeLP_ZC(G);
true

```

The Zassenhaus Conjecture for the first group in the above example, i.e. `SmallGroup(96,65)`, was proved by A. Herman and G. Singh [HS15]. They did not do all the computations for `SmallGroup(160,13)`.

Example

```

gap> C1 := CharacterTable(SymmetricGroup(5));
CharacterTable( Sym( [ 1 .. 5 ] ) )
gap> HeLP_ZC(C1);
#I The Brauer tables for the following primes are not available: [ 2, 3, 5 ].

```



```
#I ZC can't be solved, using the given data, for the orders: [ 4, 6 ].
false
gap> C2 := CharacterTable("S5");
CharacterTable( "A5.2" )
gap> HeLP_ZC(C2);
true
```

This example demonstrates the advantage of using the GAP character table library: Since GAP can't compute the Brauer tables from the ordinary table of  $S_5$  in the current implementation, they are not used in the first calculation. But in the second calculation HeLP\_ZC accesses the Brauer tables from the library and can prove the Zassenhaus Conjecture for this group, see [Her07], Section 5. This example might of course change as soon as GAP will be able to compute the needed Brauer tables.

Example

```
gap> C := CharacterTable("M11");
CharacterTable( "M11" )
gap> HeLP_ZC(C);
#I ZC can't be solved, using the given data, for the orders: [ 4, 6, 8 ].
false
gap> HeLP_sol[12];
[ ]
gap> HeLP_PrintSolution(8);
Solutions for elements of order 8:
[ [ u^4, u^2, u ],
  [ [ "2a" ], [ "2a", "4a" ], [ "2a", "4a", "8a", "8b" ] ],
  [ ---, ---, --- ],
  [ [ 1 ], [ 0, 1 ], [ 0, 0, 0, 1 ] ],
  [ [ 1 ], [ 0, 1 ], [ 0, 0, 1, 0 ] ],
  [ [ 1 ], [ 0, 1 ], [ 0, 2, -1, 0 ] ],
  [ [ 1 ], [ 0, 1 ], [ 0, 2, 0, -1 ] ],
  [ [ 1 ], [ 2, -1 ], [ 0, 0, 0, 1 ] ],
  [ [ 1 ], [ 2, -1 ], [ 0, 0, 1, 0 ] ],
  [ [ 1 ], [ 2, -1 ], [ 0, 2, -1, 0 ] ],
  [ [ 1 ], [ 2, -1 ], [ 0, 2, 0, -1 ] ] ]
```

Comparing this example to the result in [BK07a] one sees, that the existence of elements of order 12 in  $V(\mathbb{Z}M_{11})$  may not be eliminated using only the HeLP method. This may be done however by applying also the Wagner test, cf. Section 5.4 and the example for the function HeLP\_WagnerTest (3.7.1).

This example also demonstrates, why also the partial augmentations of the powers of  $u$  must be stored (and not only the partial augmentations of  $u$ ). To prove that all elements of order 8 in  $V(\mathbb{Z}M_{11})$  are rationally conjugate to group elements, it is not enough to prove that all elements  $u$  of order 8 in  $V(\mathbb{Z}M_{11})$  have all partial augmentations 1 and 0, as the fifth and sixth possibility from above still could exist in  $V(\mathbb{Z}M_{11})$ , which would not be rationally conjugate to group elements.

## 2.2 Prime Graph Question

This function checks whether the Prime Graph Question ((PQ) for short, cf. Section 5.1) can be verified using the HeLP method with the data available in GAP.

### 2.2.1 HeLP\_PQ

▷ `HeLP_PQ(OrdinaryCharacterTable/Group)` (function)

**Returns:** `true` if (PQ) can be solved using the given data, `false` otherwise

`HeLP_PQ` checks whether an affirmative answer for the Prime Graph Question for the given group can be obtained using the HeLP method, the Wagner restrictions and the data available. The argument of the function can be either an ordinary character table or a group. In the second case it will first calculate the corresponding ordinary character table. If the group in question is solvable, the Prime Graph Question has an affirmative answer by a result of W. Kimmerle and the function will return `true` without performing any calculations.

If the group is non-solvable, the ordinary character table and all  $p$ -Brauer tables for primes  $p$  for which the group is not  $p$ -solvable and which are available in GAP will be used to produce as many constraints on the torsion units as possible. Additionally, the Wagner test is applied to the results, cf. Section 5.4. In case the information suffices to obtain an affirmative answer for the Prime Graph Question, the function will return `true` and it will return `false` otherwise. Let  $p$  and  $q$  be distinct primes such that there are elements of order  $p$  and  $q$  in  $G$  but no elements of order  $pq$ . Then for  $k$  being  $p$ ,  $q$  or  $pq$  the function will save the possible partial augmentations for elements of order  $k$  and its (non-trivial) powers in `HeLP_sol[k]`. The function also does not use the previously computed partial augmentations for elements of these orders but will overwrite the content of `HeLP_sol`. If you do not like the last fact, please use `HeLP_AllOrdersPQ` (3.3.2).

Example

```
gap> C := CharacterTable("A7");
CharacterTable( "A7" )
gap> HeLP_PQ(C);
true
```

The Prime Graph Question for the alternating group of degree 7 was first proved by M. Salim [Sal11].

Example

```
gap> C := CharacterTable("L2(19)");
CharacterTable( "L2(19)" )
gap> HeLP_PQ(C);
true
gap> HeLP_ZC(C);
#I For the following orders ZC can not be solved, using the given data: [ 10 ].
false
gap> HeLP_sol[10];
[ [ [ 1 ], [ 0, 1 ], [ 0, -1, 1, 0, 1 ] ],
  [ [ 1 ], [ 0, 1 ], [ 0, 0, 0, 1, 0 ] ],
  [ [ 1 ], [ 1, 0 ], [ 0, 0, 0, 0, 1 ] ],
  [ [ 1 ], [ 1, 0 ], [ 0, 1, -1, 1, 0 ] ] ]
```

The HeLP method provides an affirmative answer to the Prime Graph Question for the group `L2(19)`, although the method doesn't solve the Zassenhaus Conjecture for that group, as there are two sets of possible partial augmentations for units of order 10 left, which do not correspond to elements which are rationally conjugate to group elements. The Zassenhaus Conjecture for this group is proved in [BM14].

Example

```
gap> C1 := CharacterTable(PSL(2,7));
CharacterTable( Group([ (3,7,5)(4,8,6), (1,2,6)(3,4,8) ]) )
```

```

gap> HeLP_PQ(C1);
#I The Brauer tables for the following primes are not available: [ 2, 3, 7 ].
#I PQ can't be solved, using the given data, for the orders: [ 6 ].
false
gap> C2 := CharacterTable("L2(7)");
CharacterTable( "L3(2)" )
gap> HeLP_PQ(C2);
true

```

This example demonstrates the advantage of using tables from the GAP character table library: Since GAP can not compute the Brauer tables corresponding to C1 they are not used in the first calculation. But in the second calculation HeLP\_PQ accesses the Brauer tables from the library and can prove the Prime Graph Question for this group, see [Her07], Section 6. This example might change, as soon as GAP will be able to compute the Brauer tables needed.

Example

```

gap> SetInfoLevel(HeLP_Info,2);
gap> C := CharacterTable("A6");
CharacterTable( "A6" )
gap> HeLP_PQ(C);
#I Checking order 2.
#I Checking order 3.
#I Checking order 5.
#I Checking order 6.
#I Checking order 10.
#I Checking order 15.
#I PQ can't be solved, using the given data, for the orders: [ 6 ].
false
gap> SetInfoLevel(HeLP_Info,1);

```

The Prime Graph Question can not be confirmed for the alternating group of degree 6 with the HeLP-method. This group is handled in [Her08c] by other means.

Example

```

gap> C := CharacterTable("L2(49)");
CharacterTable( "L2(49)" )
gap> HeLP_PQ(C);
#I The Brauer tables for the following primes are not available: [ 7 ].
#I (PQ) can't be solved, using the given data, for the orders: [ 10, 15 ].
false

```

This example shows the limitations of the program. Using the Brauer table for the prime 7 one can prove (PQ) for PSL(2,49), but this data is not available in GAP at the moment. The fact that there are no torsion units of order 10 and 15 was proved in [Her07], Proposition 6.7. See also the example in Section 4.5.

## Chapter 3

# Further functions

A short remark is probably in order on the three global variables the package is using: `HeLP_CT`, `HeLP_sol` and `HeLP_settings`. The first one stores the character table for which the last calculations were performed, the second one containing at the  $k$ 's spot the already calculated admissible partial augmentations of elements of order  $k$  (and its powers  $u^d$  for  $d \neq k$  a divisor of  $k$ ). If a function of the HeLP-package is called with a character table different from the one saved in `HeLP_CT` then the package tries to check if the character tables belong to the same group. This can be done in particular for tables from the ATLAS. If this check is successful the solutions already written in `HeLP_sol` are kept, otherwise this variable is reset. For a more detailed account see Sections 4.2, 5.2 and `HeLP_ChangeCharKeepSols` (3.4.1). In most situations, the user does not have to worry about this, the program will take care of it as far as possible. `HeLP_settings` is a variable which is used to store some settings of the program.

### 3.1 Checks for specific orders

#### 3.1.1 HeLP\_WithGivenOrder

▷ `HeLP_WithGivenOrder(CharacterTable/ListOfClassFunctions, ord)` (function)

**Returns:** List of admissible partial augmentations

Calculates the admissible partial augmentations for elements of order `ord` using only the data given in the first argument. The first argument can be an ordinary character table, a Brauer table, or a list of class functions, all having the same underlying character table. This function only uses the constraints of the HeLP method (from the class functions given), but does not apply the Wagner test 5.4. If the constraints allow only a finite number of solutions, these lists will be written in `HeLP_sol[ord]`. If for divisors  $d$  of `ord` solutions are already calculated and stored in `HeLP_sol[d]`, these will be used, otherwise the function `HeLP_WithGivenOrder` will first be applied to this order and the data given in the first argument.

Example

```
gap> C := CharacterTable("A5");
CharacterTable( "A5" )
gap> HeLP_WithGivenOrder(C, 5);
#I Number of solutions for elements of order 5: 2; stored in HeLP_sol[5].
[ [ [ 0, 1 ] ], [ [ 1, 0 ] ] ]
gap> HeLP_PrintSolution(5);
Solutions for elements of order 5:
[ [ u ],
```

```
[ [ "5a", "5b" ] ],
[      --- ],
[      [ 0, 1 ] ],
[      [ 1, 0 ] ] ]
```

Tests which partial augmentations for elements of order 5 are admissible.

Example

```
gap> C := CharacterTable("A6");
CharacterTable( "A6" )
gap> HeLP_WithGivenOrder(C, 4);
#I Number of solutions for elements of order 4: 4; stored in HeLP_sol[4].
[ [ [ 1 ], [ -1, 2 ] ], [ [ 1 ], [ 2, -1 ] ], [ [ 1 ], [ 1, 0 ] ],
  [ [ 1 ], [ 0, 1 ] ] ]
gap> HeLP_sol[4];
[ [ [ 1 ], [ -1, 2 ] ], [ [ 1 ], [ 2, -1 ] ], [ [ 1 ], [ 1, 0 ] ],
  [ [ 1 ], [ 0, 1 ] ] ]
```

Two of the non-trivial partial augmentations can be eliminated by using the Brauer table modulo the prime 3:

Example

```
gap> HeLP_WithGivenOrder(C mod 3, 4);
#I Number of solutions for elements of order 4: 2; stored in HeLP_sol[4].
[ [ [ 1 ], [ 1, 0 ] ], [ [ 1 ], [ 0, 1 ] ] ]
```

When using HeLP\_ZC also the last remaining non-trivial partial augmentation disappears, as this function applies the Wagner test, cf. 5.4 and HeLP\_WagnerTest (3.7.1):

Example

```
gap> HeLP_ZC(C);
#I ZC can't be solved, using the given data, for the orders: [ 6 ].
false
gap> HeLP_sol[4]; HeLP_sol[6];
[ [ [ 1 ], [ 0, 1 ] ] ]
[ [ [ 1 ], [ 0, 1 ], [ -2, 2, 1 ] ], [ [ 1 ], [ 1, 0 ], [ -2, 1, 2 ] ] ]
```

The following example demonstrates how one can use lists of characters to obtain constraints for partial augmentations:

Example

```
gap> C := CharacterTable("L2(49).2_1");
CharacterTable( "L2(49).2_1" )
gap> HeLP_WithGivenOrder(Irr(C){[2]}, 14);
#I The given data admit infinitely many solutions for elements of order 14.
gap> HeLP_WithGivenOrder(Irr(C){[44]}, 14);
#I The given data admit infinitely many solutions for elements of order 14.
gap> HeLP_WithGivenOrder(Irr(C){[2,44]}, 14);
#I Number of solutions for elements of order 14: 0; stored in HeLP_sol[14].
[ ]
```

Brauer tables can provide more restrictions on partial augmentations of certain torsion units:

## Example

```

gap> C := CharacterTable("J1");
CharacterTable( "J1" )
gap> HeLP_WithGivenOrder(C, 6);
#I Number of solutions for elements of order 6: 73; stored in HeLP_sol[6].
gap> B := C mod 11;
BrauerTable( "J1", 11 )
gap> HeLP_WithGivenOrder(B, 6);
#I Number of solutions for elements of order 6: 6; stored in HeLP_sol[6].
gap> HeLP_WithGivenOrder(Irr(B){[2,3]}, 6);
#I Number of solutions for elements of order 6: 6; stored in HeLP_sol[6].
gap> HeLP_PrintSolution(6);
Solutions for elements of order 6:
[ [      u^3,      u^2,      u ],
  [ [ "2a" ], [ "3a" ], [ "2a", "3a", "6a" ] ],
  [      ---,      ---,      --- ],
  [ [ 1 ], [ 1 ], [ -2, 0, 3 ] ],
  [ [ 1 ], [ 1 ], [ 2, 0, -1 ] ],
  [ [ 1 ], [ 1 ], [ 0, 0, 1 ] ],
  [ [ 1 ], [ 1 ], [ -4, 3, 2 ] ],
  [ [ 1 ], [ 1 ], [ 0, 3, -2 ] ],
  [ [ 1 ], [ 1 ], [ -2, 3, 0 ] ] ]

```

The result of the previous example can be found in [BJK11].

When dealing with many variables using lists of characters instead of a complete character table might also speed up the calculations a lot, see Section 4.3.

## Example

```

gap> C := CharacterTable("L2(27)");
CharacterTable( "L2(27)" )
gap> HeLP_WithGivenOrder(C,7);
#I Number of solutions for elements of order 7: 78; stored in HeLP_sol[7].
gap> SetInfoLevel(HeLP_Info,4);
gap> HeLP_WithGivenOrder(C,3*7);
#I Solutions for order 3 not yet calculated. Restart for this order.
#I Number of solutions for elements of order 21: 0; stored in HeLP_sol[21].
[ ]
gap> SetInfoLevel(HeLP_Info,1);

```

HeLP\_WithGivenOrder often needs to consider many cases. Set the info class HeLP\_Info to a level 4 or higher to keep track of the progress, see Section 4.4 on info levels.

### 3.1.2 HeLP\_WithGivenOrderAndPA

▷ HeLP\_WithGivenOrderAndPA(*CharacterTable/ListOfClassFunctions*, *ord*, *partaugs*)  
(function)

**Returns:** List of admissible partial augmentations

Calculates the admissible partial augmentations for elements of order *ord* using only the data given in the first argument. The first argument can be an ordinary character table, a Brauer table, or a list of class functions, all having the same underlying character table. The function uses the partial augmentations for the powers  $u^d$  with  $d$  divisors of  $k$  different from 1 and  $k$  given in *partaugs*. Here, the  $d$ 's have to be in a descending order (i.e. the orders of the  $u^d$ 's are ascending). This function

only uses the constraints of the HeLP method, but does not apply the Wagner test 5.4. Note that this function will not affect HeLP\_sol.

Example

```
gap> G := SmallGroup(48,33);; StructureDescription(G);
"SL(2,3) : C2"
gap> C := CharacterTable(G);;
gap> HeLP_WithGivenOrder(C, 4);;
#I Number of solutions for elements of order 4: 4; stored in HeLP_sol[4].
gap> HeLP_WithGivenOrder(C, 6);;
#I Number of solutions for elements of order 6: 2; stored in HeLP_sol[6].
gap> HeLP_sol[4]; HeLP_sol[6];
[[ [ 1, 0 ], [ 0, 1, 0, 0, 0 ] ], [ [ 1, 0 ], [ 0, 0, 0, 0, 1 ] ],
 [ [ 1, 0 ], [ 0, 0, 0, 1, 0 ] ], [ [ 1, 0 ], [ 0, 0, 1, 0, 0 ] ] ]
[[ [ 1, 0 ], [ 0, 1 ], [ 0, 0, 0, 0, 1, 0 ] ],
 [ [ 1, 0 ], [ 1, 0 ], [ 0, 0, 0, 0, 0, 1 ] ] ]
gap> HeLP_WithGivenOrderAndPA(C, 12, [ [ 1, 0 ], [ 0, 1 ], [ 0, 0, 0, 0, 1 ],
> [ 0, 0, 0, 0, 1, 0 ] ]);
#I Number of solutions for elements of order 12 with these partial augmentation
s for the powers: 1.
[[ [ 1, 0 ], [ 0, 1 ], [ 0, 0, 0, 0, 1 ], [ 0, 0, 0, 0, 1, 0 ],
 [ 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0, 0 ] ] ]
gap> HeLP_WithGivenOrderAndPA(C, 12, [ [ 1, 0 ], [ 0, 1 ], [ 0, 0, 0, 1, 0 ],
> [ 0, 0, 0, 0, 1, 0 ] ]);
#I Number of solutions for elements of order 12 with these partial augmentation
s for the powers: 0.
[ ]
```

In the calls of HeLP\_WithGivenOrderAndPA the function uses the following partial augmentations:

- [ 1, 0 ] for the element  $u^6$  of order 2,
- [ 0, 1 ] for the element  $u^4$  of order 3,
- [ 0, 0, 0, 0, 1 ] and [ 0, 0, 0, 1, 0 ] for the element  $u^3$  of order 4 respectively,
- [ 0, 0, 0, 0, 1, 0 ] for the element  $u^2$  of order 6.

### 3.1.3 HeLP\_WithGivenOrderAndPAAndSpecificSystem

▷ HeLP\_WithGivenOrderAndPAAndSpecificSystem(list, ord, partaugs[, b]) (function)

**Returns:** List of admissible partial augmentations

Calculates the admissible partial augmentations for elements of order *ord* using only the data given in the first argument. The first argument is a list, which can contains as entries characters or pairs with first entry a character and second entrie an integer or a mixture of these. The first argument is understood as follows: If a character  $\chi$  is not given in a pair all inequalities obtainable by this character are used. If it is given in a pair with the integer  $m$  the inequalities obtainable from the multiplicity of  $E(ord)$  taken to the power  $m$  as an eigenvalue of a representation affording  $\chi$  are used. The function uses the partial augmentations for the powers  $u^d$  with  $d$  divisors of  $k$  different from 1 and  $k$  given in *partaugs*. Here, the  $d$ 's have to be in a descending order (i.e. the orders of the  $u^d$ 's are ascending). This function only uses the constraints of the HeLP method, but does not apply the Wagner test 5.4. Note that this function will not affect HeLP\_sol.

Example

```
gap> C := CharacterTable("A5");
CharacterTable( "A5" )
gap> chi := Irr(C)[2];; psi := Irr(C)[4];
Character( CharacterTable( "A5" ), [ 4, 0, 1, -1, -1 ] )
gap> HeLP_WithGivenOrderAndPAAndSpecificSystem([[chi, 1], [chi, 2]],
> 5, [ ], true);
[[ [ [ 0, 1 ] ], [ [ 1, 0 ] ] ], [ [ -3/5, 2/5 ], [ 2/5, -3/5 ] ], [ 3/5, 3/5 ] ]
gap> sol5 := HeLP_WithGivenOrderAndPAAndSpecificSystem([[chi, 1], [chi, 2]],
> 5, [ ]);
[[ [ [ 0, 1 ] ], [ [ 1, 0 ] ] ]
```

The inequalities in the above examples are:

$$-\frac{3}{5}\varepsilon_{5a}(u) + \frac{2}{5}\varepsilon_{5b}(u) + \frac{3}{5} \in \mathbb{Z}_{\geq 0} \quad \text{and} \quad \frac{2}{5}\varepsilon_{5a}(u) + \frac{-3}{5}\varepsilon_{5b}(u) + \frac{3}{5} \in \mathbb{Z}_{\geq 0}.$$

Continuing the above example:

Example

```
gap> HeLP_WithGivenOrderAndPAAndSpecificSystem([psi],
> 2*5, [[1], sol5[1][1]], true);
[[ [ [ 0, -2/5, -2/5 ], [ 0, -1/10, -1/10 ] ], [ 0, 1/10, 1/10 ] ],
[ 0, -1/10, -1/10 ], [ 0, 1/10, 1/10 ], [ 0, 2/5, 2/5 ],
[ 0, 1/10, 1/10 ], [ 0, -1/10, -1/10 ], [ 0, 1/10, 1/10 ],
[ 0, -1/10, -1/10 ], [ 0, 1/2, 1/2, 1/2, 0, 1/2, 1/2, 1/2 ] ]
gap> HeLP_WithGivenOrderAndPAAndSpecificSystem([psi, 0], [psi, 2], [psi, 5],
> 2*5, [[1], sol5[2][1]], true);
[[ [ [ 0, -2/5, -2/5 ], [ 0, 1/10, 1/10 ] ], [ 0, 2/5, 2/5 ] ], [ 0, 1/2, 0 ] ]
```

## 3.2 Checks for specific orders with s-constant characters

When considering elements of order  $st$  (in absence of elements of this order in the group ; in particular when trying to prove (PQ)) and there are several conjugacy classes of elements of order  $s$ , it might be useful to consider  $s$ -constant characters (cf. Section 5.5) to reduce the computational complexity.

### 3.2.1 HeLP\_WithGivenOrderSConstant

▷ `HeLP_WithGivenOrderSConstant(CharacterTable/ListOfClassFunctions, s, t)` (function)

**Returns:** List of admissible "partial augmentations" or "infinite"

Calculates the admissible partial augmentations for elements  $u$  of order  $s*t$  using only the  $s$ -constant class functions that are contained in the first argument. The first argument can be an ordinary character table, a Brauer table, or a list of class functions, all having the same underlying character table.  $s$  and  $t$  have to be different prime numbers, such that there are elements of order  $s$  and  $t$  in the group, but no elements of order  $s*t$ .

The function filters which class functions given in the first argument are constant on all conjugacy classes of elements of order  $s$ . For the element  $u^s$  of order  $t$  the partial augmentations given in `HeLP_sol[t]` are used. If they are not yet calculated, the function calculates them first, using the data given in the first argument and stores them in `HeLP_sol[t]`. This function only uses the constraints



of the HeLP method, but does not apply the Wagner test 5.4. If these calculations allow an infinite number of solutions of elements of order  $st$  the function returns "infinite", otherwise it returns the finite list of solutions for elements of order  $s*t$ . The first entry of every solution is a list of the partial augmentations of  $u^s$  and the second entry is a list of the "partial augmentations" for  $u$ : the first entry of this list is the sum of the partial augmentations on all classes of elements of order  $s$  and the other entries are the partial augmentations on the classes of order  $t$ . Only in the case that the existence of units of order  $s*t$  can be excluded by this function the variable `HeLP_sol[s*t]` will be affected and `HeLP_sol[s*t]` will be set to `[]`.

Example

```
gap> C := CharacterTable("A6");;
gap> HeLP_WithGivenOrder(C, 6);
#I Number of solutions for elements of order 6: 2; stored in HeLP_sol[6].
[ [ [ 1 ], [ 0, 1 ], [ -2, 2, 1 ] ], [ [ 1 ], [ 1, 0 ], [ -2, 1, 2 ] ] ]
gap> HeLP_WithGivenOrderSConstant(C, 2, 3);
[ [ [ 0, 1 ], [ -2, 2, 1 ] ], [ [ 1, 0 ], [ -2, 1, 2 ] ] ]
gap> HeLP_WithGivenOrderSConstant(C, 3, 2);
[ [ [ 1 ], [ 3, -2 ] ] ]
```

Example

```
gap> C := CharacterTable("Sz(8)");;
gap> SetInfoLevel(HeLP_Info, 4);
gap> HeLP_WithGivenOrderSConstant(C, 7, 13);
#I Partial augmentations for elements of order 13 not yet calculated. Restart
for this order.
#I Number of non-trivial 7-constant characters in the list: 7.
[ ]
gap> SetInfoLevel(HeLP_Info, 1);
```

The last example can also be checked by using all characters in `C`, but this takes notably longer.

Example

```
gap> C := CharacterTable("Sz(32)");
CharacterTable( "Sz(32)" )
gap> L := Filtered(OrdersClassRepresentatives(C), x-> x = 31);; Size(L);
15 # I.e. HeLP_WithGivenOrder(C,31) would take hopelessly long
gap> HeLP_WithGivenOrderSConstant(C mod 2, 31, 5);
[ ]
gap> IsBound(HeLP_sol[31]);
false
```

We still have no clue about elements of order 31, but there are none of order  $5*31$ .

### 3.2.2 HeLP\_AddGaloisCharacterSums

▷ `HeLP_AddGaloisCharacterSums(CT)`

(function)

**Returns:** List of characters

Given an ordinary character table `CT` the function calculates the orbits under the action of the Galois group and returns a list of characters containing the ones contained in `CT` and the ones obtained by summing up the Galois-orbits.

### 3.3 Checks for all orders

#### 3.3.1 HeLP\_AllOrders

▷ HeLP\_AllOrders(*CharacterTable/Group*) (function)

**Returns:** true if (ZC) can be solved using the given data, false otherwise

This function does almost the same as HeLP\_ZC (2.1.1). It checks whether the Zassenhaus Conjecture can be verified for a group, but does not compute the partial augmentations of elements of order  $k$ , if HeLP\_sol[k] already exists. Thus some precalculations using e.g. HeLP\_WithGivenOrder (3.1.1) are respected. In contrast to HeLP\_ZC (2.1.1) this function also does not check whether the group is nilpotent to use the Weiss-result to have an immediate positive solution for (ZC).

This function is interesting if one wants to save time or possesses some information, which was not obtained using this package and was entered manually into HeLP\_sol.

Example

```
gap> C := CharacterTable("L2(7)");
CharacterTable( "L3(2)" )
gap> HeLP_WithGivenOrder(C,6);
#I Number of solutions for elements of order 6: 1; stored in HeLP_sol[6].
[ [ [ 1 ], [ 1 ], [ -2, 3 ] ] ]
gap> HeLP_AllOrders(C);
#I (ZC) can't be solved, using the given data, for the orders: [ 6 ].
false
gap> HeLP_ZC(C);
true
```

#### 3.3.2 HeLP\_AllOrdersPQ

▷ HeLP\_AllOrdersPQ(*CharacterTable/Group*) (function)

**Returns:** true if (PQ) can be solved using the given data, false otherwise

This function does almost the same as HeLP\_PQ (2.2.1). It checks whether the Prime Graph Question can be verified for a group, but does not compute the partial augmentations of elements of order  $k$ , if HeLP\_sol[k] already exists. Thus some precalculations using e.g. HeLP\_WithGivenOrder (3.1.1) are respected. In contrast to HeLP\_PQ (2.2.1) this function also does not check whether the group is solvable to use the Kimmerle-result to have an immediate positive solution for (ZC).

This function is interesting if one wants to save time or possesses some information, which was not obtained using this package and was entered manually into HeLP\_sol.

Example

```
gap> C := CharacterTable("A12");
CharacterTable( "A12" )
gap> HeLP_WithGivenOrder(Irr(C){[2, 4, 7]}, 2);
#I Number of solutions for elements of order 2: 37; stored in HeLP_sol[2].
gap> HeLP_WithGivenOrderSConstant(C mod 3,11,2);
[ ]
gap> HeLP_WithGivenOrder(Irr(C mod 2){[2, 3, 4, 6]}, 3);
#I Number of solutions for elements of order 3: 99; stored in HeLP_sol[3].
gap> HeLP_WithGivenOrderSConstant(C mod 2, 11, 3);
[ ]
gap> HeLP_AllOrdersPQ(C);
true
```

Thus the Prime Graph Question holds for the alternating group of degree 12. Just using `HeLP_PQ(C)` would take hopelessly long.

## 3.4 Changing the used Character Table

### 3.4.1 HeLP\_ChangeCharKeepSols

▷ `HeLP_ChangeCharKeepSols(CT)` (function)

**Returns:** nothing

This function changes the used character table to the character table `CT` and keeps all the solutions calculated so far. It is in this case the responsibility of the user that the tables belong to the same group and the ordering of the conjugacy classes in `CT` is consistent with the one in the previously used table. This function can be used to change from one table of the group to another, e.g. from a Brauer table to the ordinary table if the calculations will involve  $p$ -singular elements. (In case the involved character tables come from the ATLAS and their `InfoText` begins with "origin: ATLAS of finite groups", this is done automatically by the program.) A user may also use characters, which are normally not accessible in GAP.

To keep track of the change of the character tables one can set `HeLP_Info` to level 5. In this first example it is not realized that the character tables belong to the same group, so the solutions for elements of order 2 are recalculated (they have been reset, as another character table is used).

Example

```
gap> SetInfoLevel(HeLP_Info, 5);
gap> C := CharacterTable(SymmetricGroup(4));
CharacterTable( Sym( [ 1 .. 4 ] ) )
gap> HeLP_WithGivenOrder(C mod 2, 3);
#I USED CHARACTER TABLE CHANGED TO BrauerTable( SymmetricGroup( [ 1 .. 4 ] ), 2
), ALL GLOBAL VARIABLES RESET.
#I Number of solutions for elements of order 3: 1; stored in HeLP_sol[3].
[ [ [ 1 ] ] ]
gap> HeLP_WithGivenOrder(C, 2*3);
#I USED CHARACTER TABLE CHANGED TO CharacterTable( SymmetricGroup( [ 1 .. 4 ] )
), ALL GLOBAL VARIABLES RESET.
#I Solutions for order 2 not yet calculated. Restart for this order.
#I Solutions for order 3 not yet calculated. Restart for this order.
#I Number of solutions for elements of order 6: 0; stored in HeLP_sol[6].
[ ]
```

The recalculations of the solutions can be avoided by calling `HeLP_ChangeCharKeepSols` before using another character table.

Example

```
gap> D := CharacterTable(SymmetricGroup(4));
CharacterTable( Sym( [ 1 .. 4 ] ) )
gap> HeLP_WithGivenOrder(D mod 2, 3);
#I USED CHARACTER TABLE CHANGED TO BrauerTable( SymmetricGroup( [ 1 .. 4 ] ), 2
), ALL GLOBAL VARIABLES RESET.
#I Number of solutions for elements of order 3: 1; stored in HeLP_sol[3].
[ [ [ 1 ] ] ]
gap> HeLP_ChangeCharKeepSols(D);
#I WARNING: Change used character table without checking if the character table
s have the same underlying groups and the ordering of the conjugacy classes are
```

```

the same!
gap> HeLP_WithGivenOrder(D, 2*3);
#I Using same character table as until now; all known solutions kept.
#I      Solutions for order 2 not yet calculated. Restart for this order.
#I Number of solutions for elements of order 6: 0; stored in HeLP_sol[6].
[ ]

```

When using tables from the ATLAS this is done automatically:

Example

```

gap> CA := CharacterTable("A5");
CharacterTable( "A5" )
gap> HeLP_WithGivenOrder(CA mod 2, 5);
#I USED CHARACTER TABLE CHANGED TO BrauerTable( "A5", 2 ), ALL GLOBAL VARIABLES
  RESET.
#I Testing possibility 1 out of 1.
#I Number of solutions for elements of order 5: 2; stored in HeLP_sol[5].
[ [ [ 0, 1 ] ], [ [ 1, 0 ] ] ]
gap> HeLP_WithGivenOrder(CA, 2*5);
#I Using character table of the same group; all known solutions kept.
#I      Solutions for order 2 not yet calculated. Restart for this order.
#I Number of solutions for elements of order 10: 0; stored in HeLP_sol[10].
[ ]
gap> SetInfoLevel(HeLP_Info, 1);

```

### 3.4.2 HeLP\_Reset

▷ HeLP\_Reset() (function)

**Returns:** nothing

This function deletes all the values calculated so far and resets the global variables HeLP\_CT and HeLP\_CT to their initial value `[ [ [1] ] ]` and `CharacterTable(SmallGroup(1,1))` respectively.

## 3.5 Influencing how the Systems of Inequalities are solved

HeLP uses currently two external programs (i.e. programs that are not part of the GAP-system): `zsolve` from `4ti2` to solve the systems of linear inequalities and `redund` from `lrslib` to simplify the inequalities before handing them over to `zsolve` (HeLP can also be used without `lrslib` installed, but in general it is recommended to have `lrslib` installed). The following functions can be used to influence the behaviour of these external programmes.

### 3.5.1 HeLP\_Change4ti2Precision

▷ HeLP\_Change4ti2Precision(*string*) (function)

**Returns:** nothing

The function changes the maximum precision of the calculations of `4ti2` to solve the occurring systems of linear inequalities. The possible arguments are `"32"`, `"64"` and `"gmp"`. After calling the function the new precision will be used until this function is used again. The default value is `"32"`. A higher precision causes slower calculations. But this function might be used to increase the precision of `4ti2`, when one gets an error message like `"Error, 4ti2 Error: Results were near maximum precision"`.

(32bit). Please restart with higher precision!" stating that the results were close to the maximum 4ti2-precision.

### 3.5.2 HeLP\_UseRedund

▷ `HeLP_UseRedund(bool)`

(function)

**Returns:** nothing

The function determines whether HeLP uses 'redund' from the lrslib-package to remove redundant equations from the HeLP system. If `bool` is true 'redund' will be used in all calculation that follow, if it is false, 'redund' will not be used (which might take significantly longer). If 'redund' was not found by GAP a warning will be printed and the calculations will be performed without 'redund'. As default 'redund' will be used in all calculations.

Sometimes it is desirable to perform calculations without redund (even if it is installed and in many cases improves the performance of the package) or with a higher precision. For example, determining the partial augmentations for units of order 14 for `SmallGroup(392, 30)` involves very long calculations (when called with redund and precision 32) or cause errors (when called without redund and precision 32). However, the following works in a reasonable time.

Example

```
gap> C := CharacterTable(SmallGroup(392,30));
CharacterTable( <pc group of size 392 with 5 generators> )
gap> HeLP_UseRedund(false);
The calculations will be performed without using 'redund' from now on.
gap> HeLP_ZC(C);
Error, 4ti2 Error:
Results were near maximum precision (32bit).
Please restart with higher precision!
If you continue, your results might be wrong called from
4ti2Interface_zsolve_equalities_and_inequalities(
  [ ListWithIdenticalEntries( Size( T[1] ), 1 ) ], [ 1 ], temp[1], - temp[2]
) called from
HeLP_TestSystemINTERNAL( W[1], W[2], k, arg[3] ) called from
HeLP_WithGivenOrderAndPAINTERNAL( C, k, pa ) called from
HeLP_WithGivenOrderINTERNAL( Irr( T ), k ) called from
<function "HeLP_ZC">( <arguments> )
  called from read-eval loop at line 19 of *stdin*
you can 'quit;' to quit to outer loop, or
you can 'return;' to continue
gap> brk> quit;
#I Options stack has been reset
gap> HeLP_Change4ti2Precision("64");
The calculations of 4ti2 will be performed with precision 64 from now on.
gap> HeLP_ZC(C);
true
```

## 3.6 Checking solutions, calculating and checking solutions

### 3.6.1 HeLP\_VerifySolution

▷ `HeLP_VerifySolution(CharacterTable/ListOfClassFunctions, k[, list_paraugs])`

(function)

**Returns:** List of admissible partial augmentations

This function checks which of the partial augmentations for elements of order  $k$  given in `HeLP_sol[k]` or the optional third argument `list_paraugs` fulfill the HeLP equations obtained from the characters in the first argument. This function does not solve any inequalities, but only checks, if the given partial augmentations fulfill them. It is for this reason often faster than e.g. `HeLP_WithGivenOrder` (3.1.1).

If there is no third argument given, i.e. the augmentations from `HeLP_sol[k]` are used, the result overwrites `HeLP_sol[k]`.

Example

```
gap> C := CharacterTable("A6");;
gap> HeLP_WithGivenOrder(C, 4);
#I Number of solutions for elements of order 4: 4; stored in HeLP_sol[4].
[ [ [ 1 ], [ -1, 2 ] ], [ [ 1 ], [ 2, -1 ] ], [ [ 1 ], [ 1, 0 ] ],
  [ [ 1 ], [ 0, 1 ] ] ]
gap> HeLP_VerifySolution(C mod 3, 4);
[ [ [ 1 ], [ 1, 0 ] ], [ [ 1 ], [ 0, 1 ] ] ]
gap> HeLP_sol[4];
[ [ [ 1 ], [ 1, 0 ] ], [ [ 1 ], [ 0, 1 ] ] ]
```

Example

```
gap> C := CharacterTable("S12");;
gap> HeLP_WithGivenOrder(Irr(C mod 5){[2..6]}, 2);;
#I Number of solutions for elements of order 2: 563; stored in HeLP_sol[2].
gap> HeLP_VerifySolution(C mod 5, 2);
gap> Size(HeLP_sol[2]);
387
gap> HeLP_VerifySolution(C mod 3, 2);
gap> Size(HeLP_sol[2]);
324
```

Using `HeLP_WithGivenOrder(C mod 5, 2)` or `HeLP_WithGivenOrder(C mod 3, 2)` takes much longer since in that case a bigger system of inequalities must be solved.

### 3.6.2 HeLP\_FindAndVerifySolution

▷ `HeLP_FindAndVerifySolution(CharacterTable/ListOfClassFunctions, k)` (function)

**Returns:** List of admissible partial augmentations or "infinite"

This function provides the same functionality as `HeLP_WithGivenOrder` (3.1.1) but instead of constructing the corresponding system with all characters from the first argument `CharacterTable/ListOfClassFunctions` it does it consecutively with larger sets of characters from the argument until a finite list of solutions is found and then applies `HeLP_VerifySolution` (3.6.1) to these solutions with the entirety of the class functions in the first argument.

This function is sometimes faster than `HeLP_WithGivenOrder` (3.1.1), but the output is the same, thus the examples from `HeLP_WithGivenOrder` (3.1.1) also apply here.

### 3.6.3 HeLP\_PossiblePartialAugmentationsOfPowers

▷ `HeLP_PossiblePartialAugmentationsOfPowers(n)` (function)

**Returns:** List of partial augmentations of powers.

This function provides the possible partial augmentations of the powers of units of a given order  $n$ , if the partial augmentations of units of order  $n/p$  have been already computed for all primes  $p$  dividing  $n$ . The possibilities are sorted in the same way as, if the order  $n$  is checked with any other function like e.g. `HeLP_WithGivenOrder` (3.1.1) or `HeLP_ZC` (2.1.1). Thus, if the `InfoLevel` is high enough and one obtains that the computation of some possibility is taking too long, one can check it using `HeLP_WithGivenOrderAndPA` (3.1.2).

Example

```
gap> SetInfoLevel(HeLP_Info,4);
gap> C := CharacterTable(SmallGroup(160,91));
CharacterTable( <pc group of size 160 with 6 generators> )
gap> HeLP_WithGivenOrder(C,4);
#I      Solutions for order 2 not yet calculated. Restart for this order.
#I      Number of solutions for elements of order 4: 22; stored in HeLP_sol[4].
gap> HeLP_WithGivenOrder(C,10);
#I      Solutions for order 5 not yet calculated. Restart for this order.
#I      Number of solutions for elements of order 10: 6; stored in HeLP_sol[10].
gap> LP := HeLP_PossiblePartialAugmentationsOfPowers(20);
gap> HeLP_WithGivenOrderAndPA(Irr(C){[2..20]},20,LP[1]);
#I      Number of solutions for elements of order 20 with these partial augmentations
        for the powers: 0.
[ ]
```

## 3.7 The Wagner test

### 3.7.1 HeLP\_WagnerTest

▷ `HeLP_WagnerTest( $k$ [,  $list\_paraugs$ ,  $OrdinaryCharacterTable$ ])` (function)

**Returns:** List of admissible partial augmentations

This function applies the Wagner test (cf. Section 5.4) to the given data. If only the order  $k$  is given as argument, the Wagner test will be applied to the solutions stored in `HeLP_sol[k]`. If the arguments are the order  $k$ , a list of possible solutions  $list\_paraugs$  and an ordinary character table  $OrdinaryCharacterTable$  it applies the test to the solutions given in  $list\_paraugs$  and using the number of conjugacy classes for elements a divisor of  $k$ , which will be extracted from the head of  $OrdinaryCharacterTable$ .

Example

```
gap> C := CharacterTable("M11");
CharacterTable( "M11" )
gap> HeLP_WithGivenOrder(C,8);
#I      Number of solutions for elements of order 8: 36; stored in HeLP_sol[8].
gap> HeLP_sol[8] := HeLP_WagnerTest(8);
gap> Size(HeLP_sol[8]);
24
```

Thus the Wagner-Test eliminates 12 possible partial augmentations for elements of order 8. Continuing the example:

Example

```
gap> HeLP_WithGivenOrder(C,12);
#I      Number of solutions for elements of order 12: 7; stored in HeLP_sol[12].
[ [ [ 1 ], [ 1 ], [ 2, -1 ], [ 0, 3, -2 ], [ 1, 0, -1, 1 ] ],
```

```

[ [ 1 ], [ 1 ], [ 1, 0 ], [ 0, 3, -2 ], [ 0, 0, 0, 1 ] ],
[ [ 1 ], [ 1 ], [ -1, 2 ], [ 0, 3, -2 ], [ 0, 0, 2, -1 ] ],
[ [ 1 ], [ 1 ], [ 0, 1 ], [ 0, 3, -2 ], [ 1, 0, 1, -1 ] ],
[ [ 1 ], [ 1 ], [ 0, 1 ], [ 0, 3, -2 ], [ -1, 0, 1, 1 ] ],
[ [ 1 ], [ 1 ], [ 1, 0 ], [ 0, -3, 4 ], [ 0, 0, 0, 1 ] ],
[ [ 1 ], [ 1 ], [ -1, 2 ], [ 0, -3, 4 ], [ 1, 0, -1, 1 ] ] ]
gap> HeLP_sol[12] := HeLP_WagnerTest(12);
[ [ [ 1 ], [ 1 ], [ 1, 0 ], [ 0, 3, -2 ], [ 0, 0, 0, 1 ] ],
  [ [ 1 ], [ 1 ], [ -1, 2 ], [ 0, 3, -2 ], [ 0, 0, 2, -1 ] ],
  [ [ 1 ], [ 1 ], [ 1, 0 ], [ 0, -3, 4 ], [ 0, 0, 0, 1 ] ] ]
gap> HeLP_sol[4] := HeLP_WagnerTest(4);
gap> HeLP_WithGivenOrder(C,12);
#I Number of solutions for elements of order 12: 3; stored in HeLP_sol[12].
[ [ [ 1 ], [ 1 ], [ 2, -1 ], [ 0, 3, -2 ], [ 1, 0, -1, 1 ] ],
  [ [ 1 ], [ 1 ], [ 0, 1 ], [ 0, 3, -2 ], [ 1, 0, 1, -1 ] ],
  [ [ 1 ], [ 1 ], [ 0, 1 ], [ 0, 3, -2 ], [ -1, 0, 1, 1 ] ] ]
gap> HeLP_sol[12] := HeLP_WagnerTest(12);
[ ]

```

Thus there are no normalized units of order 12 in the integral group ring of  $M_{11}$ .

Example

```

gap> C := CharacterTable("M22");
CharacterTable( "M22" )
gap> HeLP_WagnerTest(12, [ [ [1], [1], [1,0], [0,0,1], [-3,3,2,3,-4] ] ],C);
[ ]

```

This example is taken from the appendix of [BKL08].

Sometimes the Wagner-Test may even prove the Zassenhaus Conjecture:

Example

```

gap> G := SmallGroup(96,187);
<pc group of size 96 with 6 generators>
gap> C := CharacterTable(G);
CharacterTable( <pc group of size 96 with 6 generators> )
gap> HeLP_WithGivenOrder(C,4);
#I Number of solutions for elements of order 4: 34; stored in HeLP_sol[4].
gap> HeLP_WagnerTest(4);
[ [ [ 0, 0, 1, 0, 0 ], [ 0, 0, 0, 0, 0, 0, 0, 0, 0, 1 ] ],
  [ [ 0, 0, 1, 0, 0 ], [ 0, 0, 0, 0, 0, 0, 0, 1, 0, 0 ] ],
  [ [ 0, 1, 0, 0, 0 ], [ 0, 0, 0, 0, 0, 1, 0, 0, 0, 0 ] ],
  [ [ 0, 1, 0, 0, 0 ], [ 0, 0, 0, 0, 0, 0, 0, 1, 0, 0 ] ] ]
gap> HeLP_ZC(C);
true

```

## 3.8 Output

### 3.8.1 HeLP\_PrintSolution

▷ HeLP\_PrintSolution( [k] )

(function)

**Returns:** nothing

This function prints the possible solutions in a pretty way. If a positive integer  $k$  as argument is given, then it prints the admissible partial augmentations of units of order  $k$ , if they are already



calculated. If no argument is given, the function prints information on all orders for which there are already informations.

Example

```
gap> C := CharacterTable("A5");;
gap> HeLP_ZC(C);
true
gap> HeLP_PrintSolution();
Solutions for elements of order 2:
[ [      u ],
  [ [ "2a" ] ],
  [      --- ],
  [ [ 1 ] ] ]
Solutions for elements of order 3:
[ [      u ],
  [ [ "3a" ] ],
  [      --- ],
  [ [ 1 ] ] ]
Solutions for elements of order 5:
[ [      u ],
  [ [ "5a", "5b" ] ],
  [      --- ],
  [ [ 0, 1 ] ],
  [ [ 1, 0 ] ] ]
There are no admissible partial augmentations for elements of order 6.
There are no admissible partial augmentations for elements of order 10.
There are no admissible partial augmentations for elements of order 15.
There are no admissible partial augmentations for elements of order 30.
gap> C := CharacterTable("A6");;
gap> HeLP_ZC(C);
#I ZC can't be solved, using the given data, for the orders: [ 6 ].
false
gap> HeLP_PrintSolution(6);
Solutions for elements of order 6:
[ [      u^3,      u^2,      u ],
  [ [ "2a" ], [ "3a", "3b" ], [ "2a", "3a", "3b" ] ],
  [      ---,      ---,      --- ],
  [ [ 1 ], [ 0, 1 ], [ -2, 2, 1 ] ],
  [ [ 1 ], [ 1, 0 ], [ -2, 1, 2 ] ] ]
```

## 3.9 Eigenvalue multiplicities and character values

### 3.9.1 HeLP\_MultiplicitiesOfEigenvalues

▷ HeLP\_MultiplicitiesOfEigenvalues(*chi*, *k*, *paraugs*)

(function)

**Returns:** a list of multiplicities of eigenvalues

The returned list contains at the  $l$ -th spot the multiplicity of  $E(k)^{(1-1)}$  as eigenvalue of a unit  $u$  of order  $k$  under the representation corresponding to *chi* having the partial augmentations *paraugs* for the elements  $u^d$  for divisors  $d$  different from  $k$ .

### 3.9.2 HeLP\_CharacterValue

▷ `HeLP_CharacterValue(chi, k, paraug)`

(function)

**Returns:** the character value  $\chi(u)$

The function returns the character value  $\chi(u)$  of an element  $u$  of order  $k$  having the partial augmentations  $\text{paraug}$ .

Example

```
gap> C := CharacterTable("A6");;
gap> HeLP_WithGivenOrder(C, 6);
#I Number of solutions for elements of order 6: 2; stored in HeLP_sol[6].
[ [ [ 1 ], [ 0, 1 ], [ -2, 2, 1 ] ], [ [ 1 ], [ 1, 0 ], [ -2, 1, 2 ] ] ]
gap> chi := Irr(C)[2];; # a character of degree 5
gap> HeLP_MultiplicitiesOfEigenvalues(chi, 6, HeLP_sol[6][2]);
[ 1, 0, 1, 2, 1, 0 ]
gap> HeLP_CharacterValue(chi, 6, HeLP_sol[6][2][3]);
-2
gap> HeLP_CharacterValue(chi, 6, [-2,1,2]);
-2
gap> HeLP_CharacterValue(chi, 6, [-2,2,1]);
1
```

These eigenvalues were computed manually by M. Hertweck and may be found in [Her08c].

## Chapter 4

# Extended examples

We will give some more extended examples which are intended to give the user an idea of the behaviour on different inputs, how to use the package more efficiently, to use characters not available in libraries and how InfoLevels can be helpful. We will give some more examples which are intended to give the user an idea of the behavior on different inputs and the variable HeLP\_sol. We also give hints how to use the package more efficiently, to use characters not available in libraries and how InfoLevels can be helpful.

### 4.1 The Character Table Library

Example

```
gap> G := SL(2,7);
SL(2,7)
gap> HeLP_ZC(G);
#I The Brauer tables for the following primes are not available: [ 2, 3, 7 ].
#I (ZC) can't be solved, using the given data, for the orders: [ 8 ].
false
gap> C1 := CharacterTable(G);
CharacterTable( SL(2,7) )
gap> HeLP_ZC(C1);
#I The Brauer tables for the following primes are not available: [ 2, 3, 7 ].
#I (ZC) can't be solved, using the given data, for the orders: [ 8 ].
false
gap> C2 := CharacterTable("2.L2(7)");
CharacterTable( "2.L3(2)" )
gap> HeLP_ZC(C2);
true
```

Note that the first and the second call of HeLP\_ZC (2.1.1) are equivalent – the only difference being that in the first version the character table of the group is also calculated by the function, in the second version the calculations are performed with the given character table. For the third call of HeLP\_ZC (2.1.1) a character table for SL2(7) is used which comes from the character table library. The different result is due to the fact, that in the third version the Brauer tables are available (the Brauer table for the prime  $p = 7$  is needed to rule out some non-trivial partial augmentations for elements of order 8), whereas for the first and the second call no Brauer tables are available in GAP.

## 4.2 The behavior of the variable HeLP\_sol

This sections demonstrates when the global variable HeLP\_sol is reset. This is the case if calculations are performed using (the character table of) another group than before:

Example

```
gap> C := CharacterTable("A5");
CharacterTable( "A5" )
gap> HeLP_ZC(C);
true
gap> HeLP_sol;
[ [ [ [ 1 ] ] ], [ [ [ 1 ] ] ], [ [ [ 1 ] ] ],,
  [ [ [ 0, 1 ] ], [ [ 1, 0 ] ] ], [ ],,,, [ ],,,,,, [ ],,,,,, [ ]
]
gap> C := CharacterTable("L3(7).2");
CharacterTable( "L3(7).2" )
gap> HeLP_WithGivenOrderAndPA(Irr(C){[3,7,9,10]},21,[1],[3,9,-11]);
#I Number of solutions for elements of order 21 with these partial augmentation
s for the powers: 1.
[ [ [ 1 ], [ 3, 9, -11 ], [ -6, 0, 3, 4 ] ] ]
gap> HeLP_sol;
[ [ [ [ 1 ] ] ] ]
```

The function HeLP\_WithGivenOrderAndPA (3.1.2) does not write a result in HeLP\_sol[k] (as it does not calculate all possible solutions of order  $k$ ). However HeLP\_sol is reset as a different character table is used. We continue the above example.

Example

```
gap> HeLP_WithGivenOrder(C,3);
#I Number of solutions for elements of order 3: 1; stored in HeLP_sol[3].
[ [ [ 1 ] ] ]
gap> HeLP_sol;
[ [ [ [ 1 ] ] ],, [ [ [ 1 ] ] ] ]
```

If HeLP detects that the table used belongs to the same group, HeLP\_sol is not reset:

Example

```
gap> HeLP_WithGivenOrder(C mod 7, 19);
#I Number of solutions for elements of order 19: 3; stored in HeLP_sol[19].
[ [ [ 0, 0, 1 ] ], [ [ 0, 1, 0 ] ], [ [ 1, 0, 0 ] ] ]
gap> HeLP_sol;
[ [ [ [ 1 ] ] ],, [ [ [ 1 ] ] ],,,,,,
  [ [ [ 0, 0, 1 ] ], [ [ 0, 1, 0 ] ], [ [ 1, 0, 0 ] ] ] ]
# the previously calaculated result for order 3 is still there.
```

HeLP can detect that the character tables belong to the same group, if they are identical objects in GAP or if both are tables of the same group from the ATLAS and their InfoText begins with "origin: ATLAS of finite groups" (which is usually the case for ATLAS tables). If the program can verify that the character table which is used at the current call of a function belongs to the same group as in the previous call of a function, the solutions stored in HeLP\_sol are kept. If the character table belongs to another group or it can not be made sure that the character tabel belongs to the same group, HeLP\_sol is reset to the initial value  $\begin{bmatrix} [1] \end{bmatrix}$  representing the trivial solution for units of order 1.

Not resetting HeLP\_sol can also be achieved using HeLP\_ChangeCharKeepSols (3.4.1). However, caution should be exercised when using this command since it may manipulate HeLP\_sol into something meaningless.

## Example

```

gap> G := PSL(2,7);
Group([ (3,7,5)(4,8,6), (1,2,6)(3,4,8) ])
gap> HeLP_ZC(G);
#I The Brauer tables for the following primes are not available: [ 2, 3, 7 ].
#I (ZC) can't be solved, using the given data, for the orders: [ 6 ].
false
gap> HeLP_sol;
[ [ [ [ 1 ] ] ], [ [ [ 1 ] ] ], [ [ [ 1 ] ] ], [ [ [ 1 ], [ 0, 1 ] ] ],,
  [ [ [ 1 ], [ 1 ], [ -2, 3 ] ] ], [ [ [ 0, 1 ] ], [ [ 1, 0 ] ] ],,,,, [ ],,
  [ ],,,,,, [ ],,,,,, [ ],,,,,, [ ],,,,,, [ ],,,,,, [ ],,,,,, [ ],,,,,,
  ,,,,,, [ ] ]
gap> C := CharacterTable("L2(7)") mod 7;
BrauerTable( "L3(2)", 7 )
gap> HeLP_ChangeCharKeepSols(C); #This table belongs to the same group.
gap> HeLP_WithGivenOrder(C,6);
#I Number of solutions for elements of order 6: 0; stored in HeLP_sol[6].
[ ]
gap> HeLP_sol;
[ [ [ [ 1 ] ] ], [ [ [ 1 ] ] ], [ [ [ 1 ] ] ], [ [ [ 1 ], [ 0, 1 ] ] ],,
  [ ], [ [ [ 0, 1 ] ], [ [ 1, 0 ] ] ],,,,, [ ],, [ ],,,,,, [ ],,,,,,
  [ ],,,,,, [ ],,,,,, [ ],,,,,, [ ],,,,,, [ ],,,,,, [ ],,,,,, [ ] ]
gap> C := CharacterTable("L3(7).2") mod 7;
BrauerTable( "L3(7).2", 7 )
gap> HeLP_ChangeCharKeepSols(C); #This table is from a different group
gap> HeLP_WithGivenOrder(C,19);
#I Number of solutions for elements of order 19: 3; stored in HeLP_sol[19].
[ [ [ 0, 0, 1 ] ], [ [ 0, 1, 0 ] ], [ [ 1, 0, 0 ] ] ]
gap> HeLP_sol;
[ [ [ [ 1 ] ] ], [ [ [ 1 ] ] ], [ [ [ 1 ] ] ], [ [ [ 1 ], [ 0, 1 ] ] ],,
  [ ], [ [ [ 0, 1 ] ], [ [ 1, 0 ] ] ],,,,, [ ],, [ ],,,,,,
  [ [ [ 0, 0, 1 ] ], [ [ 0, 1, 0 ] ], [ [ 1, 0, 0 ] ] ],, [ ],,,,,, [ ],,,,,,
  ,,,,,, [ ],,,,,, [ ] ]
# The content of HeLP_sol does not have a mathematical value anymore.

```

The following functions manipulate the variable `HeLP_sol`: `HeLP_ZC` (2.1.1), `HeLP_PQ` (2.2.1), `HeLP_WithGivenOrder` (3.1.1), `HeLP_WithGivenOrderSConstant` (3.2.1) (for elements of order  $t$  and if the existence of elements of order  $st$  can be excluded also for this order), `HeLP_AllOrders` (3.3.1), `HeLP_AllOrdersPQ` (3.3.2), `HeLP_VerifySolution` (3.6.1) (if existing solutions were checked), `HeLP_FindAndVerifySolution` (3.6.2). Note that the functions only will write results in `HeLP_sol[k]` if  $k$  is a divisor of the exponent of the group as this information is enough to decide whether (ZC) and (PQ) are valid for the group in consideration. In all other cases an empty list will be returned but no value will be written in `HeLP_sol[k]`.

### 4.3 Saving time

The most time consuming operation when using the functions of this package is solving the system of inequalities given by the HeLP method, see Section 5.3. This package uses the program 4ti2 to do this and it is not completely clear to the authors of this package which input is solved faster by 4ti2. In any case it is helpful to reduce the number of variables, using e.g.  $p$ -constant characters, and in many

situations it is useful to reduce the number of inequalities, i.e. of used characters.

To measure the time a function needs we use `IO_gettimeofday` from the IO-package rather than functions like `time` or `Runtime`, since these measure only the GAP time, but do not return the time the functions spend using 4ti2. We used the following function (which is essentially due to Alexander Konovalov) to measure the time used for the computation:

Example

```
TimeFunction := function(f, args)
# input: the function of which the computing time should be measured
#         and the list of arguments for this function
# output: time needed for the calculations in seconds
local start;
start := IO_gettimeofday();
CallFuncList(f,args);
return IO_gettimeofday().tv_sec - start.tv_sec;
end;
```

All times will be given in seconds. The computations were performed on an 1,6 GHz kernel.

A lot of time might be saved by testing only a few characters instead of a whole character table:

Example

```
gap> C := CharacterTable("L2(49)");
gap> TimeFunction(HeLP_WithGivenOrder, [C,35]);
#I Number of solutions for elements of order 35: 0; stored in HeLP_sol[35].
7
# I.e.: The computation took 7 seconds.
gap> TimeFunction(HeLP_FindAndVerifySolution, [C,35]);
#I Number of solutions for elements of order 35: 0; stored in HeLP_sol[35].
1
gap> TimeFunction(HeLP_FindAndVerifySolution, [Irr(C){[2]}, 35]);
#I Number of solutions for elements of order 35: 0; stored in HeLP_sol[35].
1
```

I.e.: Using only one character instead of all of them is about seven times faster in this situation and this is also quickly found by `HeLP_FindAndVerifySolution`.

Using only a few characters might even be a life saver:

Example

```
gap> C := CharacterTable("L4(3).2^2");
CharacterTable( "L4(3).2^2" )
gap> HeLP_WithGivenOrder(C, 3);
#I Number of solutions for elements of order 3: 63; stored in HeLP_sol[3].
gap> HeLP_WithGivenOrder(C, 13);
#I Number of solutions for elements of order 13: 198; stored in HeLP_sol[13].
gap> SetInfoLevel(HeLP_Info,4);
gap> TimeFunction(HeLP_WithGivenOrder, [Irr(C){[5,11,16]}, 39]);
#I Number of solutions for elements of order 39: 0; stored in HeLP_sol[39].
1301
gap> HeLP_UseRedund(false);
gap> TimeFunction(HeLP_WithGivenOrder, [Irr(C){[5,11,16]}, 39]);
#I Number of solutions for elements of order 39: 0; stored in HeLP_sol[39].
670
```

The command `HeLP_WithGivenOrder(C,39)`; started at the same time on another kernel was at possibility 1228 out of 12474 when the above example was already finished. Started without `redund` it was at possibility 5 out of 12474 and is probably still computing while you read this.

If you want to check, if the solutions you found using only a few characters satisfy the constraints of all characters, you can use `HeLP_VerifySolution` (3.6.1):

Example

```
gap> C := CharacterTable("L3(7).2");
CharacterTable( "L3(7).2" )
gap> HeLP_WithGivenOrder(C,7);;
#I Number of solutions for elements of order 7: 153; stored in HeLP_sol[7].
gap> SetInfoLevel(HeLP_Info,4);
gap> TimeFunction(HeLP_WithGivenOrder, [Irr(C){[3,7,9,10]}, 21]);
#I Number of solutions for elements of order 21: 416; stored in HeLP_sol[21].
2757
gap> Set(Filtered(OrdersClassRepresentatives(C), x -> IsPrime(x)));
[ 2, 3, 7, 19 ] # The primes dividing the order of the group
gap> HeLP_VerifySolution(C mod 2, 21);;
gap> HeLP_VerifySolution(C mod 19, 21);;
gap> Size(HeLP_sol[21]);
416
```

Thus the computed partial augmentations are best possible using HeLP. Compare the time to:

Example

```
gap> TimeFunction(HeLP_WithGivenOrder, [C, 21]);
#I Number of solutions for elements of order 21: 416; stored in HeLP_sol[21].
3460
```

Sometimes it is helpful to look at groups containing the group of interest:

Example

```
gap> C := CharacterTable("2F4(2)");
CharacterTable( "2F4(2)" )
gap> HeLP_WithGivenOrder(C, 13);;
#I Number of solutions for elements of order 13: 316; stored in HeLP_sol[13].
gap> HeLP_WithGivenOrder(C, 3);;
#I Number of solutions for elements of order 3: 1; stored in HeLP_sol[3].
gap> TimeFunction(HeLP_WithGivenOrder, [C, 39]);
#I Number of solutions for elements of order 39: 0; stored in HeLP_sol[39].
1946
gap> C:=CharacterTable("2F4(2)'.2");
CharacterTable( "2F4(2)'.2" )
gap> TimeFunction(HeLP_WithGivenOrder, [C, 39]);
#I Number of solutions for elements of order 39: 0; stored in HeLP_sol[39].
1
```

This is also a good example to use  $p$ -constant characters:

Example

```
gap> C:=CharacterTable("2F4(2)");
CharacterTable( "2F4(2)" )
gap> TimeFunction(HeLP_WithGivenOrderSConstant, [C, 13, 3]);
#I Partial augmentations for elements of order 3 not yet calculated. Restart
for this order.
#I Number of non-trivial 13-constant characters in the list: 19.
0
```

For some groups switching redund on and off gives major improvements.

Example

```
gap> C := CharacterTable(SmallGroup(160,91));
CharacterTable( <pc group of size 160 with 6 generators> )
gap> TimeFunction(HeLP_ZC, [C]);
41
```

Without redund HeLP\_ZC(C) ran for over 400 hours without a result.

Example

```
gap> C := CharacterTable(SmallGroup(96,12));
CharacterTable( <pc group of size 96 with 6 generators> )
gap> HeLP_UseRedund(false);
gap> TimeFunction(HeLP_ZC, [C]);
6
```

Running this example using redund the computations does not proceed for elements of order 12.

## 4.4 Using InfoLevels

HeLP provides different InfoLevels for different situations. The variable controlling the InfoLevel is HeLP\_Info and it might be changed using SetInfoLevel(HeLP\_Info, n) to set the InfoLevel to n. The maximal HeLP\_Info entry is 5, the default InfoLevel is 1. The examples below give some idea, how one can use HeLP\_Info, but do not give complete information on all possibilities.

If one is only interested whether (ZC) or (PQ) can be solved using the HeLP method, one can set HeLP\_Info to 0:

Example

```
gap> C := CharacterTable("M11");
CharacterTable( "M11" )
gap> HeLP_ZC(C);
#I ZC can't be solved, using the given data, for the orders: [ 4, 6, 8 ].
false
gap> SetInfoLevel(HeLP_Info, 0);
gap> HeLP_ZC(C);
false
```

If the InfoLevel is set to 2, the functions HeLP\_ZC (2.1.1) and HeLP\_PQ (2.2.1) print information which order of torsion units is currently considered, so that the user can keep track of the progress. This may be used for bigger groups to see, if the calculations might finish at some point. Continuing the above example:

Example

```
gap> SetInfoLevel(HeLP_Info, 2);
gap> HeLP_PQ(C);
#I Checking order 2.
#I Checking order 3.
#I Checking order 5.
#I Checking order 10.
#I Checking order 11.
#I Checking order 15.
#I Checking order 22.
#I Checking order 33.
#I Checking order 55.
true
```



HeLP\_Info at InfoLevel 3 provides also some information about the used ordinary character table or Brauer tables:

Example

```
gap> SetInfoLevel(HeLP_Info, 3);
gap> HeLP_PQ(C);
#I Checking order 2.
#I Using table BrauerTable( "M11", 3 ).
#I Checking order 3.
#I Using table BrauerTable( "M11", 3 ).
#I Using table BrauerTable( "M11", 11 ).
#I Checking order 5.
#I Using table BrauerTable( "M11", 3 ).
#I Checking order 10.
#I Using table BrauerTable( "M11", 3 ).
#I Checking order 11.
#I Using table BrauerTable( "M11", 3 ).
#I Checking order 15.
#I Using table BrauerTable( "M11", 3 ).
#I Using table BrauerTable( "M11", 11 ).
#I Checking order 22.
#I Using table BrauerTable( "M11", 3 ).
#I Checking order 33.
#I Using table BrauerTable( "M11", 3 ).
#I Using table BrauerTable( "M11", 11 ).
#I Using table BrauerTable( "M11", 2 ).
#I Checking order 55.
#I Using table BrauerTable( "M11", 3 ).
true
```

Setting HeLP\_Info to 4 is useful when there are many possibilities for the partial augmentations of the powers of some unit. A good example is the example on "L4(3).2~2" in the section on Time Saving 4.3, see above: If you see quickly that almost nothing is happening, you might want to change your strategy.

HeLP\_Info at level 5 informs the user on all changes of the used character table. Using it makes sense, if you work with the command HeLP\_ChangeCharKeepSols (3.4.1).

## 4.5 Non-standard characters

The package also allows using characters even if the whole character table is not available. E.g. induced characters:

Example

```
gap> C := CharacterTable("U3(8)");
CharacterTable( "U3(8)" )
gap> G := PSU(3,8);
<permutation group of size 5515776 with 2 generators>
gap> A := AutomorphismGroup(G);
<group of size 99283968 with 4 generators>
gap> AllCharacterTableNames(Size,Size(A));
[ "3.U3(8).6", "3.U3(8).S3" ]
```

This means: The character table of the automorphism group A of PSU(3,8) is not available in GAP. However one can use induced characters:



## 4.6 A complete example: (PQ) for the MacLaughlin simple group

This section gives a demonstration of many functions of the package. The goal is to verify the Prime Graph Question for the MacLaughlin simple group, which was proved in [BK07b]

Example

```
gap> C := CharacterTable("McL");
CharacterTable( "McL" )
gap> SetInfoLevel(HeLP_Info,4);
```

The function `HeLP_PQ(C)` would take really long. Instead one can use `HeLP_AllOrdersPQ(C)` several times on a high `InfoLevel`. Any time you see the function needs long, just try some manual calculations. Compute first the partial augmentations of elements of prime order:

Example

```
gap> HeLP_WithGivenOrder(C,2);;
#I Number of solutions for elements of order 2: 1; stored in HeLP_sol[2].
gap> HeLP_WithGivenOrder(C mod 2,3);;
#I Number of solutions for elements of order 3: 4; stored in HeLP_sol[3].
gap> HeLP_WithGivenOrder(C mod 3,5);;
#I Number of solutions for elements of order 5: 6; stored in HeLP_sol[5].
gap> HeLP_WithGivenOrder(C mod 3,7);;
#I Number of solutions for elements of order 7: 174; stored in HeLP_sol[7].
gap> HeLP_WithGivenOrder(C mod 3,11);;
#I Number of solutions for elements of order 11: 20; stored in HeLP_sol[11].
```

For mixed order in most situations  $p$ -constant characters are interesting. Check the tables for such characters of small degree.

Example

```
gap> HeLP_WithGivenOrderSConstant(Irr(C){[2,3,4,5]},7,3);
#I Number of non-trivial 7-constant characters in the list: 4.
[ ]
gap> HeLP_WithGivenOrderSConstant(Irr(C){[2,3,4,5]},11,2);
#I Number of non-trivial 11-constant characters in the list: 4.
[ ]
gap> HeLP_WithGivenOrderSConstant(Irr(C){[2,3,4,5]},11,3);
#I Number of non-trivial 11-constant characters in the list: 4.
[ ]
gap> HeLP_WithGivenOrderSConstant(Irr(C mod 3){[2,3,4,5]},7,5);
#I Number of non-trivial 7-constant characters in the list: 4.
[ ]
gap> HeLP_WithGivenOrderSConstant(Irr(C mod 3){[2,3,4,5]},7,11);
#I Number of non-trivial 7-constant characters in the list: 4.
[ ]
gap> HeLP_WithGivenOrderSConstant(Irr(C mod 3){[2,3,4,5]},11,5);
#I Number of non-trivial 11-constant characters in the list: 2.
[ ]
```

These calculations are enough to obtain an affirmative answer to the Prime Graph Question:

Example

```
gap> HeLP_AllOrdersPQ(C);
#I Checking order 2.
#I Using the known solutions for elements of order 2.
```

```
#I Checking order 3.  
#I Using the known solutions for elements of order 3.  
#I Checking order 5.  
#I Using the known solutions for elements of order 5.  
#I Checking order 7.  
#I Using the known solutions for elements of order 7.  
#I Checking order 11.  
#I Using the known solutions for elements of order 11.  
#I Checking order 21.  
#I Using the known solutions for elements of order 21.  
#I Checking order 22.  
#I Using the known solutions for elements of order 22.  
#I Checking order 33.  
#I Using the known solutions for elements of order 33.  
#I Checking order 35.  
#I Using the known solutions for elements of order 35.  
#I Checking order 55.  
#I Using the known solutions for elements of order 55.  
#I Checking order 77.  
#I Using the known solutions for elements of order 77.  
true
```

Checking these computations takes a few minutes.

## Chapter 5

# Background

### 5.1 The Zassenhaus Conjecture and the Prime Graph Question

Let  $G$  be a finite group and let  $\mathbb{Z}G$  denote its integral group ring. Let  $V(\mathbb{Z}G)$  be the group of units of augmentation one, aka. normalized units. An element of the unit group of  $\mathbb{Z}G$  is called a torsion element, if it has finite order.

A long standing conjecture of H.J. Zassenhaus asserts that every normalized torsion unit of  $\mathbb{Z}G$  is conjugate within  $\mathbb{Q}G$  ("rationally conjugate") to an element of  $G$ , see [Zas74] or [Seh93], Section 37. This is the first of his three famous conjectures about integral group rings and the only one which is nowadays still open, hence it is referred to as the Zassenhaus Conjecture (ZC). This conjecture asserts that the torsion part of the units of  $\mathbb{Z}G$  is as far determined by  $G$  as possible.

Considering the difficulty of the problem W. Kimmerle raised the question, whether the Prime Graph of the normalized units of  $\mathbb{Z}G$  coincides with that one of  $G$  (cf. [Kim07] Problem 21). This is the so called Prime Graph Question (PQ). The prime graph of a group is the loop-free, undirected graph having as vertices those primes  $p$ , for which there is an element of order  $p$  in the group. Two vertices  $p$  and  $q$  are joined by an edge, provided there is an element of order  $pq$  in the group. In the light of this description, the Prime Graph Question asks, whether there exists an element of order  $pq$  in  $G$  provided there exists an element of order  $pq$  in  $V(\mathbb{Z}G)$  for every pair of primes  $(p, q)$ .

In general, by a result of J. A. Cohn and D. Livingstone [CL65], Corollary 4.1, and a result of M. Hertweck [Her08a], the following is known about the possible orders of torsion units in integral group rings:

*Theorem:* The exponents of  $V(\mathbb{Z}G)$  and  $G$  coincide. Moreover, if  $G$  is solvable, any torsion unit in  $V(\mathbb{Z}G)$  has the same order as some element in  $G$ .

### 5.2 Partial augmentations and the structure of HeLP sol

For a finite group  $G$  and an element  $x \in G$  let  $x^G$  denote the conjugacy class of  $x$  in  $G$ . The partial augmentation with respect to  $x$  or rather the conjugacy class of  $x$  is the map  $\varepsilon_x$  sending an element  $u$  to the sum of the coefficients at elements of the conjugacy class of  $x$ , i.e.

$$\varepsilon_x: \mathbb{Z}G \rightarrow \mathbb{Z}, \quad \sum_{g \in G} z_g g \mapsto \sum_{g \in x^G} z_g.$$

Let  $u$  be a torsion element in  $V(\mathbb{Z}G)$ . By results of G. Higman, S.D. Berman and M. Hertweck the following is known for the partial augmentations of  $u$ :

*Theorem:* ([Seh93], Proposition (1.4); [Her07], Proposition 2.2)  $\varepsilon_1(u) = 0$  if  $u \neq 1$  and  $\varepsilon_x(u) = 0$  if the order of  $x$  does not divide the order of  $u$ .

Partial augmentations are connected to (ZC) and (PQ) via the following result, which is due to Z. Marciniak, J. Ritter, S. Sehgal and A. Weiss [MRSW87], Theorem 2.5:

*Theorem:* A torsion unit  $u \in V(\mathbb{Z}G)$  of order  $k$  is rationally conjugate to an element of  $G$  if and only if all partial augmentations of  $u^d$  vanish, except one (which then is necessarily 1) for all divisors  $d$  of  $k$ .

The last statement also explains the structure of the variable `HeLP_sol`. In `HeLP_sol[k]` the possible partial augmentations for an element of order  $k$  and all powers  $u^d$  for  $d$  dividing  $k$  (except for  $d = k$ ) are stored, sorted ascending w.r.t. order of the element  $u^d$ . For instance, for  $k = 12$  an entry of `HeLP_sol[12]` might be of the following form:

[ [ 1 ], [ 0, 1 ], [ -2, 2, 1 ], [ 1, -1, 1 ], [ 0, 0, 0, 1, -1, 0, 1, 0, 0 ] ].

The first sublist [ 1 ] indicates that the element  $u^6$  of order 2 has the partial augmentation 1 at the only class of elements of order 2, the second sublist [ 0, 1 ] indicates that  $u^4$  of order 3 has partial augmentation 0 at the first class of elements of order 3 and 1 at the second class. The third sublist [ -2, 2, 1 ] states that the element  $u^3$  of order 4 has partial augmentation -2 at the class of elements of order 2 while 2 and 1 are the partial augmentations at the two classes of elements of order 4 respectively, and so on. Note that this format provides all necessary information on the partial augmentations of  $u$  and its powers by the above restrictions on the partial augmentations.

For more details on when the variable `HeLP_sol` is modified or reset and how to influence this behavior see Section 4.2 and `HeLP_ChangeCharKeepSols` (3.4.1).

### 5.3 The HeLP equations

Denote by  $x^G$  the conjugacy class of an element  $x$  in  $G$ . Let  $u$  be a torsion unit in  $V(\mathbb{Z}G)$  of order  $k$  and  $D$  an ordinary representation of  $G$  over a field contained in  $\mathbb{C}$  with character  $\chi$ . Then  $D(u)$  is a matrix of finite order and thus diagonalizable over  $\mathbb{C}$ . Let  $\zeta$  be a primitive  $k$ -th root of unity, then the multiplicity  $\mu_l(u, \chi)$  of  $\zeta^l$  as an eigenvalue of  $D(u)$  can be computed via Fourier inversion and equals

$$\mu_l(u, \chi) = \frac{1}{k} \sum_{1 \neq d|k} \text{Tr}_{\mathbb{Q}(\zeta^d)/\mathbb{Q}}(\chi(u^d)\zeta^{-dl}) + \frac{1}{k} \sum_{x^G} \varepsilon_x(u) \text{Tr}_{\mathbb{Q}(\zeta)/\mathbb{Q}}(\chi(x)\zeta^{-l}).$$

As this multiplicity is a non-negative integer, we have the constraints

$$\mu_l(u, \chi) \in \mathbb{Z}_{\geq 0}$$

for all ordinary characters  $\chi$  and all  $l$ . This formula was given by I.S. Luthar and I.B.S. Passi [LP89].

Later M. Hertweck showed that it may also be used for a representation over a field of characteristic  $p > 0$  with character  $\varphi$ , if  $p$  is coprime to  $k$  [Her07], § 4. In that case one has to ignore the  $p$ -singular conjugacy classes (i.e. the classes of elements with an order divisible by  $p$ ) and the above formula becomes

$$\mu_l(u, \varphi) = \frac{1}{k} \sum_{1 \neq d|k} \text{Tr}_{\mathbb{Q}(\zeta^d)/\mathbb{Q}}(\varphi(u^d)\zeta^{-dl}) + \frac{1}{k} \sum_{x^G, p \nmid o(x)} \varepsilon_x(u) \text{Tr}_{\mathbb{Q}(\zeta)/\mathbb{Q}}(\varphi(x)\zeta^{-l}).$$

Again, as this multiplicity is a non-negative integer, we have the constraints

$$\mu_l(u, \varphi) \in \mathbb{Z}_{\geq 0}$$

for all Brauer characters  $\varphi$  and all  $l$ .

These equations allow to build a system of integral inequalities for the partial augmentations of  $u$ . Solving these inequalities is exactly what the HeLP method does to obtain restrictions on the possible values of the partial augmentations of  $u$ . Note that some of the  $\varepsilon_x(u)$  are a priori zero by the results in the above sections.

For  $p$ -solvable groups representations over fields of characteristic  $p$  can not give any new information compared to ordinary representations by the Fong-Swan-Rukolaine Theorem [CR90], Theorem 22.1.

## 5.4 The Wagner test

We also included a result motivated by a theorem R. Wagner proved 1995 in his Diplomarbeit [Wag95]. This result gives a further restriction on the partial augmentations of torsion units. We included it into the functions HeLP\_ZC (2.1.1), HeLP\_PQ (2.2.1), HeLP\_AllOrders (3.3.1), HeLP\_AllOrdersPQ (3.3.2) and HeLP\_WagnerTest (3.7.1) and call it "Wagner test".

*Theorem:* For a torsion unit  $u \in V(\mathbb{Z}G)$ , a group element  $s$ , a prime  $p$  and a natural number  $j$  we have

$$\sum_{x^{p^j} \sim s} \varepsilon_x(u) \equiv \varepsilon_s(u^{p^j}) \pmod{p}.$$

Combining the Theorem with the HeLP-method may only give new insight, if  $p^j$  is a proper divisor of the order of  $u$ . Wagner did obtain this result for  $s = 1$ , when  $\varepsilon_s(u) = 0$  by the Berman-Higman Theorem. In the case that  $u$  is of prime power order this is a result of J.A. Cohn and D. Livingstone [CL65].

## 5.5 $s$ -constant (and $(s,t)$ -constant) characters

If one is interested in units of mixed order  $s * t$  for two primes  $s$  and  $t$  (e.g. if one is studying the Prime Graph Question) an idea to make the HeLP method more efficient was introduced by V. Bovdi and A. Konovalov in [BK10], page 4. Assume one has several conjugacy classes of elements of order  $s$ , and a character taking the same value on all of these classes. Then the coefficient of every of these conjugacy classes in the system of inequalities of this character, which is obtained via the HeLP method, is the same. Also the constant terms of the inequalities do not depend on the partial augmentations of elements of order  $s$ . Thus for such characters one can reduce the number of variables in the inequalities by replacing all the partial augmentations on classes of elements of order  $s$  by their sum. To obtain the formulas for the multiplicities of the HeLP method one does not need the partial augmentations of elements of order  $s$ . Characters having the above property are called  $s$ -constant. In this way the existence of elements of order  $s * t$  can be excluded in a quite efficient way without doing calculations for elements of order  $s$ .

There is also the concept of  $(s,t)$ -constant characters, being constant on both, the conjugacy classes of elements of order  $s$  and on the conjugacy classes of elements of order  $t$ . The implementation of this is however not yet part of this package.

## 5.6 Known results about (ZC) and (PQ)

At the moment as this documentation was written, to the best of our knowledge, the following results were available for the Zassenhaus Conjecture and the Prime Graph Question:

For the Zassenhaus Conjecture only the following reduction is available:

*Theorem:* Assume the Zassenhaus Conjecture holds for a group  $G$ . Then (ZC) holds for  $G \times C_2$  [HK06], Corollary 3.3, and  $G \times \Pi$ , where  $\Pi$  denotes a nilpotent group of order prime to the order of  $G$  [Her08b], Proposition 8.1.

With this reductions in mind the Zassenhaus Conjecture is known for:

- Nilpotent groups [Wei91],
- Cyclic-By-Abelian groups [CMdR13],
- Groups containing a normal Sylow subgroup with abelian complement [Her06],
- Frobenius groups whose order is divisible by at most two different primes [JPM00],
- Groups  $X \rtimes A$ , where  $X$  is abelian,  $A$  elementary abelian and  $A$  acts faithfully and irreducibly on  $X$  [SW86],
- Groups  $X \rtimes A$ , where  $X$  and  $A$  are abelian and  $A$  is of prime order  $p$  such that  $p$  is smaller then any prime divisor of the order of  $X$  [MRSW87],
- All groups of order up to 159 and some other groups of order up to 200 listed in [HS15],
- The non-abelian simple groups  $A_5$  [LP89],  $A_6 \simeq PSL(2, 9)$  [Her08c],  $PSL(2, 7)$ ,  $PSL(2, 11)$ ,  $PSL(2, 13)$  [Her07],  $PSL(2, 8)$ ,  $PSL(2, 17)$  [KK13] / [Gil13],  $PSL(2, 19)$ ,  $PSL(2, 23)$  [BM14] and some extensions of these groups.

For the Prime Graph Question the following strong reduction was obtained in [KK13]:

*Theorem:* Assume the Prime Graph Question holds for all almost simple images of a group  $G$ . Then (PQ) also holds for  $G$ .

Here a group  $G$  is called almost simple, if it is sandwiched between the inner automorphism group and the whole automorphism group of a non-abelian simple group  $S$ . I.e.  $Inn(S) \leq G \leq Aut(S)$ . Keeping this reduction in mind (PQ) is known for:

- Solvable groups [Kim06],
- The groups  $PSL(2, p)$  where  $p$  denotes a prime [Her07],
- Half of the sporadic simple groups and their automorphism groups, for an overview see [KK13],
- The groups  $A_7, A_8$  [Sal11],  $A_9, A_{10}$  [Sal13],
- Almost simple groups whose order is divisible by at most three different primes [KK13] and [BM14]. (This implies that it holds for all groups with an order divisible by at most three primes, using the reduction result above.)



## Chapter 6

# Remarks on technical problems and the implementation

### 6.1 Making the HeLP-package run

A checklist what to do, if the package does not work:

- Make sure you have sufficiently new versions of the following software:
  - [GAP](#) (at least 4.7.4)
  - the GAP-package [CTblLib](#) (at least 1.2.2)
  - the GAP-package [IO](#) (at least 4.2; see also the next bullet point if this package can not be loaded)
  - the GAP-package [4ti2Interface](#) (at least 2015.04.29; this package needs the IO-package)Usually all these packages should come with a sufficiently recent GAP-installation (4.7.8 or newer) and should be contained in the pkg-folder of the GAP-installation. To see if they are working you can load them by typing `LoadPackage(" [name] ")`; after starting GAP, where `[name]` is the name of the package.
- The IO-package needs a C-part to be compiled. To see if this has already been done on your system, you can enter `LoadPackage("IO")`; after starting GAP. If the result is `fail` and the package is contained in the pkg-folder, than most likely the C-part is not yet compiled. For information on installation and in particular on how to compile the C-part, see the [manual](#) (in particular Chapter 2) or the README-file of that package.
- Please make sure that [www.4ti2.de](http://www.4ti2.de) (Version 1.6.5 or newer) is installed. In case of an error-message "The executable 'zsolve' provided by the software 4ti2 was not found." after typing `LoadPackage("HeLP")`; either the software is not properly installed or installed in a directory where GAP can not find it, i.e. a directory not contained in the path-variable. The content of this variable can typically be displayed by typing `echo $PATH` (Linux, Mac) `echo %PATH%` (Windows) in a terminal or a command prompt. The manual of 4ti2 contains several pages of information on how to install the program. Note that the installation of 4ti2 requires gcc (g++) and gmp installed (which come with many Linux installations or can be installed using a package manager). Make sure to execute all four commands indicated in the 4ti2 manual (possibly without the `-prefix=`-command):

```
./configure --prefix=INSTALLATION-DIRECTORY
make
make check
make install-exec
```

Depending on the settings of your system you might need root privileges (type `sudo` in front of every command) to unpack the files and install them. To check whether the installation worked, you can enter `zsolve` in a terminal. In case one of the required programs (`g++` or `gmp`) was not installed when running `make` for the first time, you might need to run `make clean` and the above commands several times to compile 4ti2 successfully. If you already have 4ti2 installed in a directory not contained in the `path`-variable and want to avoid a re-installation, in many cases the following helps:

- Start a terminal and access a path written in your `bash` or `system_bash`. Typically `usr/local/bin` should work.
- Run `ln -s [PathToZsolve] zsolve`, where `[PathToZsolve]` is the path to the executable `zsolve`. This sets a symlink to the right place. E.g. `ln -s /opt/4ti2/bin/zsolve zsolve` was used on the (Linux) computers in Stuttgart.
- We also recommend to install [lrslib](#), at least version 4.3 (note that version 4.2 or older sometimes produces unwanted behavior). This software provides the 'redund' command, which can be switched on and off within HeLP, but which often leads to better performances (cf. `HeLP_UseRedund` (3.5.2)). For installation see the User's Guide or the Readme-file on the above mentioned homepage. Usually, after unpacking in a directory contained in the `path`-variable it should be enough to call
 

```
make all
```

 (possibly as root) inside the `lrslib`-directory.
- In case this does not help to get HeLP running, please feel more than welcome to contact one of the maintainers of the package.

## 6.2 Computations with big coefficients - precision problems

When using 4ti2, HeLP calls it per default with a maximum precision of 32bit. In case one gets error messages stating that the results of the calculations were near the maximum precision, the used precision can be influenced by using `HeLP_Change4ti2Precision` (3.5.1). For groups with a "reasonable" order there should be no problems with the precision.

## 6.3 How much 4ti2 is really there?

The reason, why the program 4ti2 is used in this package, is basically that it can solve systems of linear inequalities efficiently and there exists a good GAP-Interface to it. However there is only one line of code where a function is called which accesses 4ti2. Thus the effort of using another solver of inequalities would be not so big, if there is a GAP-Interface for it. If you are aware of such a solver and would like to use it in this package, please contact the authors of this package. We will be happy to help.

# References

- [Avi] David Avis. Irslib – reverse search vertex enumeration program. Available at <http://cgm.cs.mcgill.ca/~avis/C/lrs.html>. 5
- [BJK11] V. A. Bovdi, E. Jespers, and A. B. Konovalov. Torsion units in integral group rings of Janko simple groups. *Math. Comp.*, 80(273):593–615, 2011. 14
- [BK07a] V. A. Bovdi and A. B. Konovalov. Integral group ring of the first Mathieu simple group. In *Groups St. Andrews 2005. Vol. 1*, volume 339 of *London Math. Soc. Lecture Note Ser.*, pages 237–245. Cambridge Univ. Press, Cambridge, 2007. 9
- [BK07b] V. A. Bovdi and A. B. Konovalov. Integral group ring of the McLaughlin simple group. *Algebra Discrete Math.*, (2):43–53, 2007. 35
- [BK10] V. A. Bovdi and A. B. Konovalov. Torsion units in integral group ring of Higman-Sims simple group. *Studia Sci. Math. Hungar.*, 47(1):1–11, 2010. 39
- [BKL08] V. A. Bovdi, A. B. Konovalov, and S. Linton. Torsion units in integral group ring of the Mathieu simple group  $M_{22}$ . *LMS J. Comput. Math.*, 11:28–39, 2008. 24
- [BM14] A. Bächle and L. Margolis. Rational conjugacy of torsion units in integral group rings of non-solvable groups. *preprint*, [arXiv: 1305. 7419\[math. RT\]](https://arxiv.org/abs/1305.7419), 2014. 22 pages. 10, 40
- [CL65] J. A. Cohn and D. Livingstone. On the structure of group algebras. I. *Canad. J. Math.*, 17:583–593, 1965. 37, 39
- [CMdR13] M. Caicedo, L. Margolis, and Á. del Río. Zassenhaus conjecture for cyclic-by-abelian groups. *J. Lond. Math. Soc. (2)*, 88(1):65–78, 2013. 40
- [CR90] C. W. Curtis and I. Reiner. *Methods of representation theory. Vol. I*. Wiley Classics Library. John Wiley & Sons, Inc., New York, 1990. With applications to finite groups and orders, Reprint of the 1981 original, A Wiley-Interscience Publication. 39
- [Gil13] J. Gildea. Zassenhaus conjecture for integral group ring of simple linear groups. *J. Algebra Appl.*, 12(6):1350016, 10, 2013. 40
- [Her06] M. Hertweck. On the torsion units of some integral group rings. *Algebra Colloq.*, 13(2):329–348, 2006. 40
- [Her07] M. Hertweck. Partial augmentations and Brauer character values of torion units in group rings. *Preprint*, 2007. e-print [arXiv:math.RA/0612429v2\[math.RA\]](https://arxiv.org/abs/math.RA/0612429v2). 5, 9, 11, 38, 40

- [Her08a] M. Hertweck. The orders of torsion units in integral group rings of finite solvable groups. *Comm. Algebra*, 36(10):3585–3588, 2008. [37](#)
- [Her08b] M. Hertweck. Torsion units in integral group rings of certain metabelian groups. *Proc. Edinb. Math. Soc. (2)*, 51(2):363–385, 2008. [40](#)
- [Her08c] M. Hertweck. Zassenhaus conjecture for  $A_6$ . *Proc. Indian Acad. Sci. Math. Sci.*, 118(2):189–195, 2008. [8](#), [11](#), [26](#), [40](#)
- [HK06] C. Höfert and W. Kimmerle. On torsion units of integral group rings of groups of small order. In *Groups, rings and group rings*, volume 248 of *Lect. Notes Pure Appl. Math.*, pages 243–252. Chapman & Hall/CRC, Boca Raton, FL, 2006. [8](#), [40](#)
- [HS15] A. Herman and G. Singh. Revisiting the Zassenhaus Conjecture on torsion units for the integral group rings of small groups. *To appear in Pro. Mat. Sci. Ind. Ac. Sci.*, 2015. [8](#), [40](#)
- [JPM00] S. O. Juriaans and C. Polcino Milies. Units of integral group rings of Frobenius groups. *J. Group Theory*, 3(3):277–284, 2000. [40](#)
- [Kim06] W. Kimmerle. On the prime graph of the unit group of integral group rings of finite groups. In *Groups, rings and algebras*, volume 420 of *Contemp. Math.*, pages 215–228. Amer. Math. Soc., Providence, RI, 2006. [40](#)
- [Kim07] W. Kimmerle. Mini-Workshop: Arithmetik von Gruppenringen. *Oberwolfach Reports*, 4(4):3209–3239, 2007. [37](#)
- [KK13] W. Kimmerle and A. B. Konovalov. Recent advances on torsion subgroups of integral group rings. *To appear in Proc. of Groups St Andrews 2013*, 2013. [40](#)
- [LP89] I.S. Luthar and I.B.S. Passi. Zassenhaus conjecture for  $A_5$ . *Proc. Indian Acad. Sci. Math. Sci.*, 99(1):1–5, 1989. [5](#), [7](#), [38](#), [40](#)
- [MRSW87] Z. Marciniak, J. Ritter, S. Sehgal, and A. Weiss. Torsion units in integral group rings of some metabelian groups. II. *Journal of Number Theory*, 25(3):340–352, 1987. [38](#), [40](#)
- [Sal11] M. Salim. Kimmerle’s conjecture for integral group rings of some alternating groups. *Acta Math. Acad. Paedagog. Nyházi. (N.S.)*, 27(1):9–22, 2011. [10](#), [40](#)
- [Sal13] M. Salim. The prime graph conjecture for integral group rings of some alternating groups. *Int. J. Group Theory*, 2(1):175–185, 2013. [40](#)
- [Seh93] S.K. Sehgal. *Units in integral group rings*, volume 69 of *Pitman Monographs and Surveys in Pure and Applied Mathematics*. Longman Scientific & Technical, Harlow, 1993. [37](#), [38](#)
- [Sri64] B. Srinivasan. On the modular characters of the special linear group  $SL(2, p^n)$ . *Proc. London Math. Soc. (3)*, 14:101–114, 1964. [34](#)
- [SW86] S. K. Sehgal and A. Weiss. Torsion units in integral group rings of some metabelian groups. *J. Algebra*, 103(2):490–499, 1986. [40](#)

- [tt] 4 ti 2 team. 4ti2—a software package for algebraic, geometric and combinatorial problems on linear spaces. Available at [www.4ti2.de](http://www.4ti2.de). 5
- [Wag95] R. Wagner. Zassenhausvermutung über die Gruppen  $\mathrm{PSL}(2, p)$ . Diplomarbeit Universität Stuttgart, Mai 1995. 39
- [Wei91] A. Weiss. Torsion units in integral group rings. *J. Reine Angew. Math.*, 415:175–187, 1991. 40
- [Zas74] H. Zassenhaus. On the torsion units of group rings. *Estudos de Matemática em homenagem ao Prof. A. Almeida Costa, Instituto de Alta Cultura (Portugese)*, pages 119–126, 1974. 37

# Index

HeLP, [5](#)

HeLP\_AddGaloisCharacterSums, [17](#)

HeLP\_AllOrders, [18](#)

HeLP\_AllOrdersPQ, [18](#)

HeLP\_Change4ti2Precision, [20](#)

HeLP\_ChangeCharKeepSols, [19](#)

HeLP\_CharacterValue, [26](#)

HeLP\_FindAndVerifySolution, [22](#)

HeLP\_MultiplicitiesOfEigenvalues, [25](#)

HeLP\_PossiblePartialAugmentationsOf-  
Powers, [22](#)

HeLP\_PQ, [10](#)

HeLP\_PrintSolution, [24](#)

HeLP\_Reset, [20](#)

HeLP\_UseRedund, [21](#)

HeLP\_VerifySolution, [21](#)

HeLP\_WagnerTest, [23](#)

HeLP\_WithGivenOrder, [12](#)

HeLP\_WithGivenOrderAndPA, [14](#)

HeLP\_WithGivenOrderAndPAAndSpecific-  
System, [15](#)

HeLP\_WithGivenOrderSConstant, [16](#)

HeLP\_ZC, [6](#)